

# PAPER SOFT

26

sped. in abb. post. gr. 1170

Anno - I N° 26 - 21 dicembre 1984



**Bowling**  
**Il topo ed il drago**



**Space War**



**Excalibur**  
**Breakcheese**



**Sintetizzatore di suono**



**Il paroliere**

# Guida all'input dei programmi sullo ZX Spectrum

Le "parole" comprese tra parentesi graffe indicano i caratteri grafici predefiniti (G), il tasto (numero seguente la G), la necessità di premere il tasto insieme a CAPS SHIFT (eventuale S precedente la G) e il numero di ripetizioni del tasto richieste (eventuale numero all'inizio della "parola"). I caratteri grafici definiti dal programma sono invece indicati da lettere maiuscole (corrispondenti al tasto da usare) sottolineate.

Quando leggete	Premete	Vedrete
{G1}	1	■
{G2}	2	■
{G3}	3	■
{G4}	4	■
{G5}	5	■
{G6}	6	■
{G7}	7	■
{G8}	8	■
{SG1}	CAPS SHIFT 1	■
{SG2}	CAPS SHIFT 2	■
{SG3}	CAPS SHIFT 3	■
{SG4}	CAPS SHIFT 4	■
{SG5}	CAPS SHIFT 5	■
{SG6}	CAPS SHIFT 6	■
{SG7}	CAPS SHIFT 7	■
{SG8}	CAPS SHIFT 8	■

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9

Quando leggete	Premete	Vedrete
<u>A</u>	A	Simbolo grafico definito nel programma in uso.
<u>B</u>	B	
<u>C</u>	C	
<u>D</u>	D	
<u>E</u>	E	
<u>F</u>	F	
<u>G</u>	G	
<u>H</u>	H	
<u>I</u>	I	
<u>J</u>	J	
<u>K</u>	K	
<u>L</u>	L	
<u>M</u>	M	
<u>N</u>	N	
<u>O</u>	O	
<u>P</u>	P	
<u>Q</u>	Q	
<u>R</u>	R	
<u>S</u>	S	
<u>T</u>	T	
<u>U</u>	U	

Se non siete già in modo G, entrateci schiacciando contemporaneamente CAPS SHIFT e 9

Se dovete uscire dal modo G, schiacciate 9


# Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono "parole" racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da [ < > ] deve essere

premutato contemporaneamente al tasto COMMODORE. Inoltre, se tra parentesi si trova un carattere alfabetico "solitario", questo dovrà essere premuto contemporaneamente al tasto CONTROL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME	■	{CYN}	CTRL 4	■	[ < 7 > ]	G 7	■
{HOME}	CLR/HOME	■	{PUR}	CTRL 5	■	[ < 8 > ]	G 8	■
{SU}	SHIFT ↑ CRSR ↓	■	{GRN}	CTRL 6	■	{F1}	f1	■
{GIU'}	↑ CRSR ↓	■	{BLU}	CTRL 7	■	{F2}	f2	■
{SIN}	SHIFT ← CRSR →	■	{YEL}	CTRL 8	■	{F3}	f3	■
{DES}	← CRSR →	■	[ < 1 > ]	G 1	■	{F4}	f4	■
{RVS}	CTRL 9	■	[ < 2 > ]	G 2	■	{F5}	f5	■
{OFF}	CTRL 0	■	[ < 3 > ]	G 3	■	{F6}	f6	■
{BLK}	CTRL 1	■	[ < 4 > ]	G 4	■	{F7}	f7	■
{WHT}	CTRL 2	■	[ < 5 > ]	G 5	■	{F8}	f8	■
{RED}	CTRL 3	■	[ < 6 > ]	G 6	■			■



# PAPER SOFT

- |         |   |           |   |
|---------|---|-----------|---|
| paddle  |  apple             | <b>4</b>  | <b>Bowling</b><br>di P. Case trad. e adatt. di M. Cerofolini                            |
| paddle  |  apple             | <b>7</b>  | <b>Il topo ed il drago</b><br>di T. Smith trad. e adatt. di M. Cerofolini               |
|         |  TI-99/4A          | <b>10</b> | <b>Space War</b><br>di H. Gorman trad. e adatt. di E. Re Garbagnati                     |
| 16/48K  |  Sinclair Spectrum | <b>21</b> | <b>Excalibur</b><br>di L. Howetts trad. e adatt. di C. Panzalis                         |
| 16/48K  |  Sinclair Spectrum | <b>23</b> | <b>Breakcheese</b><br>di P. & C. Mc Cabe trad. e adatt. di C. Panzalis                  |
|         | <b>C64</b>  | <b>25</b> | <b>Sintetizzatore di suono</b><br>di B. Kahiwinkle trad. e adatt. di M. Cristuib Grizzi |
| 3 K RAM |  VIC-20            | <b>27</b> | <b>Il paroliere</b><br>di B. Meyers Jr. trad. e adatt. di M. Anticoli e S. Albarelli    |

J. soft s.r.l.

**DIREZIONE, REDAZIONE,  
AMMINISTRAZIONE**

Via Rosellini, 12  
20124 MILANO  
Tel. (02) 68.88.228

**DIRETTORE RESPONSABILE:**

Pietro Dell'Orco

**COORDINAMENTO TECNICO:**

Riccardo Paolillo

**REDAZIONE:**

Lucio Bragagnolo  
Mauro Cristuib. Grizzi

**GRAFICA E IMPAGINAZIONE:**

Margherita La Noce  
Ivana Rossi  
Raffaella Toffolatti

**FOTOCOMPOSIZIONE:**

d&b Via Vignola, 5  
Tel. 02/59.85.08  
20133 MILANO

**CONTABILITÀ:**

Giulia Pedrazzini  
Flavia Bonaiti

**AUTORIZZAZIONE ALLA  
PUBBLICAZIONE:**

Tribunale di Milano n° 200  
del 14.04.1984

**STAMPA:**

Elcograf - Beverate (CO)



Rivista associata  
all'Unione stampa  
Periodica Italiana

**PUBBLICITÀ**

Concessionario per l'Italia e l'Estero Reina s.r.l.  
Via Washington, 50  
20046 MILANO  
Tel. (02) 49.88.066 (5 linee R.A.)  
Tlx. 316213 REINA I  
Concessionario esclusivo per la  
DIFFUSIONE in Italia e Estero:  
SODIP - Via Zuretti, 25  
20125 MILANO

Spedizione in abbonamento  
postale Gruppo II/70  
Prezzo della rivista L. 1.000  
Numeri arretrati L. 2.000

© TUTTI I DIRITTI DI  
RIPRODUZIONE O TRADUZIONE  
DEGLI ARTICOLI PUBBLICATI  
SONO RISERVATI

# Bowling

Con questo programma potete giocare a bowling da soli o con un amico, senza lasciare il vostro confortevole salotto, e soprattutto senza stancarvi troppo.

Innanzitutto vi viene richiesto il numero dei partecipanti, quindi il livello di difficoltà (da 1 a 3): a seconda del livello prescelto il cursore si muove più o meno velocemente, così da permettervi di prendere la mira con maggiore o

minore facilità.

Per stabilire la direzione del lancio dovete premere il bottone della paddle quando ritenete che il cursore (la vostra boccia) sia nella giusta traiettoria. Al termine di ogni lancio nella parte bassa dello schermo appare il punteggio totalizzato da ciascun giocatore nel corso della partita. Se giocate da soli ricordatevi di utilizzare la paddle 0.

```

1 DIM B(26),C(26),A(10): CALL -
  936: DEF FN R(X) = INT ( RND
  (X) * X)
2 TEXT : CALL - 936: VTAB 10: HTAB
  2: PRINT "QUANTI GIOCATORI (
  1 0 2) ";; INPUT V: IF V < 1
  OR V > 2 THEN 2
3 FOR X = 1 TO 26:B(X) = 0:C(X) =
  0: NEXT X: FOR X = 1 TO 10:A
  (X) = 0: NEXT X
4 HTAB 2: VTAB 12: PRINT "LIVELL
  0 DI DIFFICOLTA' (1-3) ";; INPUT
  T1:T1 = 4 - T1: IF T1 < 1 OR
  T1 > 3 THEN 4
5 XX = 1:H = 1: CALL - 936
6 K = 1
8 C2 = 13:C1 = 2:G = 1
9 GR : COLOR= C1: HLIN 0,39 AT 0
  : HLIN 0,39 AT 19: COLOR= C2
  : FOR X = 0 TO 39: VLIN 1,18
  AT X: NEXT X
10 T = 1
20 COLOR= C1
51 PLOT 37,5: PLOT 37,6: PLOT 37
  ,9: PLOT 37,10: PLOT 37,13: PLOT
  37,14
52 PLOT 35,7: PLOT 35,8: PLOT 35
  ,11: PLOT 35,12: PLOT 33,9: PLOT
  33,10
53 PLOT 39,3: PLOT 39,4: PLOT 39
  ,7: PLOT 39,8: PLOT 39,11: PLOT
  39,12: PLOT 39,15: PLOT 39,1
  6
54 FOR X = 1 TO 10:A(X) = 1: NEXT
  X
55 Y = 1:Z = 18:H = 1
56 COLOR= 0: FOR X = 19 TO 22: VLIN
  24,35 AT X: NEXT X:GG = G: IF
  GG > 10 THEN GG = 10
57 IF G = G5 AND K = K1 THEN C3 =
  3
58 IF K < > K1 OR G < > G5 THEN
  C3 = 15
59 IF G > 10 THEN C3 = 3
60 K1 = K:G5 = G: GOSUB 1500

```

```

61 FOR X = Y TO Z STEP SGN (Z -
  Y): COLOR= C2: PLOT 0,XX: COLOR=
  0: PLOT 0,X
63 IF PEEK ( - 16288 + K) > 127
  THEN 90
65 XX = X
66 IF T1 = 1 THEN 70
68 FOR Z1 = 1 TO (T1 - 1) ^ 3: NEXT
  Z1
70 NEXT X: GOSUB 2500:H = H + 1:
  IF (H / 2) - INT (H / 2) =
  0 THEN 77
75 Y = 1:Z = 18: GOTO 61
77 Z = 1:Y = 18: GOTO 61
90 REM
95 COLOR= C2: PLOT 0,XX
98 PLOT 0,X
99 YY = 0: FOR Y = 1 TO 39: IF SCRN(
  Y,XX) = C1 THEN 103
100 COLOR= C2: PLOT YY,XX
101 COLOR= 0: PLOT Y,XX:YY = Y
102 FOR Z1 = 1 TO (T1 - 1) ^ 3: NEXT
  Z1,Y
103 COLOR= C2: PLOT YY,XX
104 X = XX
105 IF X = 2 OR X = 3 THEN 900
106 IF X = 16 OR X = 17 THEN 902
107 IF X = 14 OR X = 15 THEN 910
108 IF X < 2 OR X > 17 THEN 200
109 IF X = 4 OR X = 5 THEN 920
110 IF X = 6 THEN 180
111 IF X < > 7 THEN 118
112 GOSUB 220: GOSUB 240: GOSUB
  280: IF A(4) = 1 AND A(2) =
  1 THEN GOSUB 270
113 IF X < 2 OR X > 17 THEN 200
114 IF X = 4 OR X = 5 THEN 920
115 IF A(4) = 1 AND FN R(2) = 1
  THEN GOSUB 270
116 IF A(2) = 1 THEN 1250
117 GOTO 200
118 IF X = 13 THEN 190

```

```

119 IF X < > 12 THEN 125
120 GOSUB 230: GOSUB 260: GOSUB
290: IF A(6) = 1 AND A(3) =
1 THEN GOSUB 300
121 IF A(6) = 1 AND FN R(2) = 1
THEN GOSUB 300
123 IF A(3) = 1 THEN 1310
124 GOTO 200
125 IF X < > 8 THEN 135
126 GOSUB 210: GOSUB 220: GOSUB
280: IF A(1) = 1 OR A(2) = 1
THEN GOSUB 270
127 IF A(1) = 1 AND FN R(4) > 0
THEN GOSUB 300
128 IF A(1) < > 1 THEN GOSUB 2
50
129 IF A(1) = 1 OR A(2) = 1 THEN
GOSUB 240
130 IF A(1) = 1 AND A(4) = 1 AND
A(8) = 1 AND FN R(3) > 0 THEN
GOSUB 290
131 IF FN R(3) = 1 AND A(1) = 1
THEN 950
132 GOTO 200
135 IF X < > 11 THEN 145
136 GOSUB 210: GOSUB 230: GOSUB
290: IF A(1) = 1 OR A(3) = 1
AND FN R(4) > 0 THEN GOSUB
300
137 IF A(1) = 1 AND FN R(4) > 0
THEN GOSUB 270
138 IF A(1) = 1 OR A(3) = 1 THEN
GOSUB 260
139 IF A(1) < > 1 THEN GOSUB 2
50
141 IF FN R(3) = 1 AND A(1) = 1
THEN 960
142 IF A(1) = 1 AND A(6) = 1 AND
A(9) = 1 AND FN R(3) > 1 THEN
GOSUB 280
144 GOTO 200
145 IF X = 10 THEN 170
147 GOSUB 210: IF FN R(6) > 0 THEN
GOSUB 250
148 GOSUB 220: GOSUB 280: IF A(1
) = 1 THEN GOSUB 230
149 IF A(1) = 1 OR A(2) = 1 THEN
970
150 IF A(1) = 1 AND FN R(6) > 0
THEN 980
151 IF A(1) = 0 AND A(5) = 1 THEN
990
152 IF A(5) = 1 AND FN R(5) > 0
THEN GOSUB 290
153 GOTO 200
170 GOSUB 210: IF FN R(6) > 0 THEN
GOSUB 250
171 GOSUB 230: GOSUB 290: IF A(1
) = 1 THEN GOSUB 220
172 IF A(1) = 1 OR A(3) = 1 THEN
1000
173 IF A(1) = 1 AND FN R(6) > 0
THEN 1010
174 IF A(1) = 0 AND A(5) = 1 THEN
1020
175 IF A(5) = 1 AND FN R(5) > 0
THEN GOSUB 280
176 GOTO 200
180 GOSUB 220: GOSUB 240: GOSUB

```

```

280: IF A(8) = 1 OR A(4) = 1
THEN GOSUB 270
181 IF A(2) = 1 THEN 1030
182 IF A(3) = 1 AND A(5) = 1 AND
FN R(3) > 0 THEN GOSUB 250

183 IF A(2) = 1 AND FN R(2) = 1
THEN GOSUB 290
184 GOTO 200
190 GOSUB 230: GOSUB 260: GOSUB
290: IF A(9) = 1 OR A(6) = 1
THEN GOSUB 300
191 IF A(3) = 1 THEN 1040
193 IF A(3) = 1 AND FN R(2) = 1
THEN GOSUB 280
200 GOSUB 2000
201 IF T < > 1 THEN 207
202 RR = 10 - J: B(G + 13 * K - 13
) = 10 - J: IF B(G + 13 * K -
13) < > 10 THEN 207
203 B(G + K * 13 - 13) = 20: X = 2
: GOTO 400
207 IF T < > 1 THEN 209
208 T = 0: GOTO 55
209 IF T = 0 THEN 1200
210 Y = 33: Z = 9: GOTO 310
220 Y = 35: Z = 7: GOTO 310
230 Y = 35: Z = 11: GOTO 310
240 Y = 37: Z = 5: GOTO 310
250 Y = 37: Z = 9: GOTO 310
260 Y = 37: Z = 13: GOTO 310
270 Y = 39: Z = 3: GOTO 310
280 Y = 39: Z = 7: GOTO 310
290 Y = 39: Z = 11: GOTO 310
300 Y = 39: Z = 15
310 COLOR= C2: PLOT Y, Z: PLOT Y,
Z + 1
311 S = - 16336
318 FOR I1 = 1 TO 3
320 LT = PEEK (S)
325 NEXT I1
350 RETURN
360 PRINT "": RETURN
400 IF G = 1 THEN 416
401 IF G = 2 THEN 405
404 IF B(G - 2 + K * 13 - 13) =
20 AND B(G - 1 + K * 13 - 13
) = 20 THEN 1100
405 D = B(G + K * 13 - 13): IF D =
20 THEN D = 10
407 IF B(G - 1 + K * 13 - 13) =
20 AND B(G + 13 * K - 13) =
20 THEN 500
410 IF B(G - 1 + 13 * K - 13) =
20 THEN 1400
415 IF B(G - 1 + 13 * K - 13) =
10 THEN 1110
416 IF B(G + 13 * K - 13) = 10 OR
B(G + 13 * K - 13) = 20 THEN
500
420 IF G < = 10 THEN 1120
440 REM
500 GG = G: IF G > 10 THEN GG = 1
0
501 B = B(G + 13 * K - 13): IF B >
9 THEN GOSUB 360
506 IF B > 10 THEN GOSUB 360

```

```

510 VTAB 2 * K + 19: HTAB G * 3:
    B = B(G + 13 * K - 13): IF B
    < 10 THEN PRINT RR;B - RR;
514 IF B = 20 THEN PRINT "X";
517 IF B = 10 THEN PRINT RR;"/"
    ;
520 IF V = 1 THEN 560
530 IF G > = 10 AND K = 1 AND C
    (10 + 13 * K - 13) = 0 THEN
    1130
540 IF I = 1 THEN 1140
550 IF G > = 10 AND K = 2 AND C
    (23) = 0 THEN 1150
560 IF C(10 + V * 13 - 13) < >
    0 THEN 600
570 IF V = 1 THEN 1150
580 IF K = 2 THEN 1170:K = 2: GOTO
    10
585 K = 2: GOTO 10
590 GOTO 10
600 FOR X = 1 TO 2000: NEXT X
602 TEXT : CALL - 936
603 PRINT
605 FOR X = 1 TO V: PRINT "PUNTI
    ";C(10 + X * 13 - 13);" PER
    GIOCATORE ";X: PRINT : PRINT
    : NEXT X
610 PRINT "PREMERE RETURN PER UN
    'ALTRO GIOCO";: INPUT A$: GOTO
    2
900 GOSUB 270: IF X = 2 AND A(7)
    = 1 THEN GOSUB 280
901 GOTO 200
902 GOSUB 300: IF X = 17 AND A(1
    0) = 1 THEN GOSUB 290
903 GOTO 200
910 GOSUB 260: GOSUB 300: IF A(6
    ) = 1 THEN 915
912 IF A(10) = 1 AND X = 14 THEN
    915
913 GOTO 200
915 GOSUB 290: GOTO 200
920 GOSUB 240: GOSUB 270: IF A(4
    ) = 1 THEN 925
922 IF A(7) = 1 AND X = 5 THEN 9
    24
923 GOTO 200
924 GOSUB 290: GOTO 200
925 GOSUB 280: GOTO 200
950 GOSUB 250: GOSUB 230: GOSUB
    260: GOSUB 290: GOTO 200
960 GOSUB 250: GOSUB 220: GOSUB
    240: GOSUB 280: GOTO 142
970 GOSUB 240: GOSUB 270: GOTO 1
    50
980 GOSUB 260: IF FN R(8) > 1 THEN
    GOSUB 300
981 GOTO 151
990 GOSUB 290: GOSUB 250: GOTO 1
    52
1000 GOSUB 260: GOSUB 300: GOTO
    173
1010 GOSUB 240: IF FN R(8) > 1 THEN
    GOSUB 270
1011 GOTO 174
1020 GOSUB 280: GOSUB 250: GOTO
    175
1030 GOSUB 250: GOSUB 260: IF A(
    5) = 1 AND FN R(4) > 1 THEN
    GOSUB 300

```

```

1031 GOTO 182
1040 GOSUB 250: GOSUB 240: IF A(
    5) = 1 AND FN R(4) > 1 THEN
    GOSUB 270
1041 GOTO 193
1100 C(G - 2 + K * 13 - 13) = C(G
    - 3 + K * 13 - 13) + 20 + R
    R
1105 VTAB 2 * K + 20: HTAB G * 3
    - 6: PRINT C(G - 2 + 13 * K
    - 13);
1106 IF B(G + 13 * K - 13) = 20 THEN
    500
1108 GOTO 405
1110 C(G - 1 + 13 * K - 13) = C(G
    - 2 + 13 * K - 13) + 10 + R
    R: IF G < 12 THEN 1115
1111 GOTO 416
1115 VTAB 2 * K + 20: HTAB G * 3
    - 3: PRINT C(G - 1 + 13 * K
    - 13);
1118 GOTO 416
1120 C(G + 13 * K - 13) = C(G - 1
    + K * 13 - 13) + B(G + 13 *
    K - 13)
1125 VTAB 2 * K + 20: HTAB G * 3
    : PRINT C(G + 13 * K - 13);
1128 GOTO 500
1130 G = G + 1:I = 1: GOTO 10
1140 G = 10:K = 2:I = 0: GOTO 10
1150 G = G + 1: GOTO 10
1170 G = G + 1:K = 1: GOTO 10
1200 B(G + 13 * K - 13) = B(G + 1
    3 * K - 13) + 10 - J - RR: IF
    B(G + 13 * K - 13) = 10 THEN
    X = 1
1205 GOTO 400
1250 GOSUB 250: IF A(5) < > 1 THEN
    200
1260 GOSUB 290: IF FN R(3) = 1 THEN
    200
1270 GOSUB 260: IF FN R(3) = 1 THEN
    200
1280 GOSUB 260: IF FN R(3) = 1 THEN
    200
1300 GOSUB 300
1301 IF FN R(3) > 0 THEN GOSUB
    260
1302 IF FN R(3) > 0 THEN GOSUB
    300
1305 GOTO 200
1310 GOSUB 250: IF A(5) < > 1 THEN
    200
1315 GOSUB 280: IF FN R(3) = 1 THEN
    200
1320 GOSUB 240: IF FN R(3) = 1 THEN
    200
1330 GOSUB 240: IF FN R(3) = 1 THEN
    200
1350 GOSUB 270
1351 IF FN R(3) > 0 THEN GOSUB
    240
1352 IF FN R(3) > 0 THEN GOSUB
    270
1355 GOTO 200
1400 C(G - 1 + K * 13 - 13) = C(G
    - 2 + K * 13 - 13) + D + 10
    : IF G > 11 THEN 415
1405 VTAB 2 * K + 20: HTAB G * 3
    - 3: PRINT C(G - 1 + 13 * K
    - 13);

```

```

1410 GOTO 416
1500 COLOR= C3: IF GG = 1 THEN 1
      550
1501 IF GG = 2 THEN 1560
1502 IF GG = 3 THEN 1570
1503 IF GG = 4 THEN 1580
1504 IF GG = 5 THEN 1590
1505 IF GG = 6 THEN 1600
1510 IF GG = 7 THEN 1610
1520 IF GG = 8 THEN 1620
1530 IF GG = 9 THEN 1640
1540 GOTO 1630
1550 GOSUB 1710: GOSUB 1720: RETURN

1560 GOSUB 1700: GOSUB 1750: GOSUB
      1760: GOSUB 1720: GOSUB 1730
      : RETURN
1570 GOSUB 1700: GOSUB 1750: GOSUB
      1740: GOSUB 1730: GOSUB 1760
      : RETURN
1580 GOSUB 1710: GOSUB 1760: GOSUB
      1740: GOSUB 1750
1585 RETURN
1590 GOSUB 1700: GOSUB 1710: GOSUB
      1760: GOSUB 1740: GOSUB 1730
      : RETURN
1600 GOSUB 1700: GOSUB 1710: GOSUB
      1720: GOSUB 1730: GOSUB 1740
      : GOSUB 1760: RETURN
1610 GOSUB 1700: GOSUB 1750: GOSUB
      1740: RETURN
1620 GOSUB 1700: GOSUB 1710: GOSUB
      1720: GOSUB 1730: GOSUB 1740
      : GOSUB 1750: GOSUB 1760: RETURN

1630 VLIN 24,35 AT 16: GOSUB 170
      0: GOSUB 1710: GOSUB 1720: GOSUB
      1730: GOSUB 1740: GOSUB 1750
      : RETURN
1640 GOSUB 1710: GOSUB 1700: GOSUB
      1750: GOSUB 1760: GOSUB 1740
      : GOSUB 1730: RETURN
1700 HLIN 19,22 AT 24: HLIN 19,2
      2 AT 25: RETURN

```

```

1710 VLIN 24,30 AT 19: RETURN
1720 VLIN 29,35 AT 19: RETURN
1730 HLIN 19,22 AT 34: HLIN 19,2
      2 AT 35: RETURN
1740 VLIN 29,35 AT 22: RETURN
1750 VLIN 24,30 AT 22: RETURN
1760 HLIN 19,22 AT 29: HLIN 19,2
      2 AT 30: RETURN
2000 J = 0: FOR A = 1 TO 10:A(A) =
      0: NEXT A
2001 FOR A = 1 TO 10
2003 IF A = 1 THEN 2020
2004 IF A = 2 THEN 2030
2005 IF A = 3 THEN 2040
2006 IF A = 4 THEN 2050
2007 IF A = 5 THEN 2060
2012 IF A = 6 THEN 2070
2013 IF A = 7 THEN 2080
2014 IF A = 8 THEN 2090
2015 IF A = 9 THEN 2100
2016 IF A = 10 THEN 2110
2020 Y = 33:Z = 9: GOTO 2200
2030 Y = 35:Z = 7: GOTO 2200
2040 Y = 35:Z = 11: GOTO 2200
2050 Y = 37:Z = 5: GOTO 2200
2060 Y = 37:Z = 9: GOTO 2200
2070 Y = 37:Z = 13: GOTO 2200
2080 Y = 39:Z = 3: GOTO 2200
2090 Y = 39:Z = 7: GOTO 2200
2100 Y = 39:Z = 11: GOTO 2200
2110 Y = 39:Z = 15: GOTO 2200
2200 IF SCRN( Y,Z) = C1 THEN 22
      10
2205 GOTO 2250
2210 A(A) = 1
2220 J = J + 1
2250 NEXT A: RETURN
2500 FOR I1 = 1 TO 2
2505 L1 = PEEK (S) + PEEK (S) +
      PEEK (S) + PEEK (S) + PEEK
      (S)
2509 NEXT I1
2510 RETURN
5000 END

```



## Il topo ed il drago

paddle

In questo gioco di caccia dovrete cercare di totalizzare il punteggio più alto possibile entro un limitato periodo di tempo.

Utilizzerete le paddle per muovere il topolino (rappresentato da un rettangolo marrone) verso il formaggio (un rettangolo giallo) e farglielo mangiare. Ogni volta che il topolino riesce nel suo intento il vostro punteggio aumenta ed un nuovo pezzo di formaggio appare in un'altra zona dello schermo. Ma, a complicare le co-

se, c'è un drago che dà la caccia al topolino, e quando riesce a catturarlo se lo mangia ponendo termine al gioco.

Via via che diverrete più abili il punteggio aumenterà ed allora sia il formaggio che il drago si muoveranno più velocemente rendendo sempre più arduo il raggiungimento dell'obiettivo. Questo gioco è particolarmente divertente perché interagiscono 4 fattori interessanti:

1) lo scopo del gioco (il punteggio aumenta ogni

volta che riuscite a far mangiare il formaggio al topolino)

2) complicazioni successive dovute al drago che vuole a sua volta mangiare il topolino

3) il fattore tempo (il tempo concesso è limitato ed inoltre la velocità di movimento aumenta)

4) la veste grafica (i simboli colorati si muovono sullo schermo).

```
1000 REM -----
-----
1010 REM IL GIOCO DEL DRAGO E D
EL TOPO
1020 REM -----
-----
1030 REM TITOLO ED ISTRUZIONI
1040 DEF FN R(X) = INT ( RND (
1) * X) + 1
1050 TEXT
1060 CALL - 936
1065 INVERSE
1070 VTAB 5
1080 PRINT "IL"
1090 PRINT "GIOCO"
1100 PRINT "DEL"
1110 PRINT "TOPO"
1120 PRINT "E DEL"
1130 PRINT "DRAGO"
1135 NORMAL
1140 GOSUB 10100
1150 POKE 32,7
1160 POKE 33,33
1170 VTAB 15
1200 GOSUB 10100
1210 VTAB 22
1220 PRINT
1230 PRINT "QUESTO E' UN GIOCO D
I CACCIA"
1240 GOSUB 10100
1250 PRINT "CHE RIGUARDA UN TOPO
, UN DRAGO,"
1260 GOSUB 10100
1270 PRINT "ED UNA QUANTITA' ILL
IMITATA DI"
1280 GOSUB 10100
1290 PRINT "FORMAGGIO."
1300 GOSUB 10100
1310 PRINT
1320 GOSUB 10100
1330 PRINT "VOI CONTROLLATE IL T
OPO (IL"
1340 GOSUB 10100
1350 PRINT "QUADRATO MARRONE) CO
N LE"
1360 GOSUB 10100
1370 PRINT "PADDLE."
1380 GOSUB 10100
1390 PRINT
1400 GOSUB 10100
1410 PRINT "IL DRAGO (CHE E' ROS
SO)"
1420 GOSUB 10100
1430 PRINT "VI DA' LA CACCIA."
1440 GOSUB 10100
1450 PRINT
1460 GOSUB 10100
1470 PRINT "IL VOSTRO OBIETTIVO
E' MANGIARE"
```

```
1480 GOSUB 10100
1490 PRINT "QUANTO PIU' FORMAGGI
O (QUADRATI"
1500 GOSUB 10100
1510 PRINT "GIALLI) POTETE MENTR
E"
1520 GOSUB 10100
1530 PRINT "SFUGGITE AL DRAGO."
1540 GOSUB 10100
1550 PRINT
1560 GOSUB 10100
1570 PRINT "AVETE 60 SECONDI PER
MANGIARE"
1580 GOSUB 10100
1590 PRINT "QUANTO PIU' POTETE."

1600 GOSUB 10100
1610 PRINT
1620 GOSUB 10100
1630 PRINT "UN AVVERTIMENTO: SE
DIVENTATE"
1640 GOSUB 10100
1650 PRINT "TROPPO BRAVI, IL FOR
MAGGIO "
1660 GOSUB 10100
1670 PRINT "COMINCIA A MUOVERSI
ED IL"
1680 GOSUB 10100
1690 PRINT "DRAGO DIVENTA PIU' A
STUTO."
1700 GOSUB 10100
1710 PRINT
1720 GOSUB 10100
1730 PRINT "SE VI SENTITE PRONTI
ADESSO"
1740 GOSUB 10100
1750 PRINT "PREMETE UNO DEI DUE
BOTTONI"
1760 GOSUB 10100
1770 PRINT "DELLE PADDLE."
1790 REM * INIZIALIZZAZIONE PRO
GRAMMA *
1800 TM = 60
1810 DN = 5
1820 DIM DX(DN),DY(DN)
1830 HS = 0
1840 REM * NUOVO GIOCO *
1850 IF PEEK ( - 16287) > 127 THEN
1900
1860 IF PEEK ( - 16286) > 127 THEN
1900
1870 GOTO 1850
1900 REM * INIZIALIZZAZIONE GIO
CO *
1910 GR
1920 CALL - 936
1930 TG = TM
1940 TK = 3
1950 VTAB 22
```



```

1960 HTAB 6
1970 PRINT "TEMPO ";TG
1980 SC = 0
1990 VTAB 22
2000 HTAB 25
2010 PRINT "PUNTI ";SC
2020 REM * CONTORNO *
2030 BG = 5
2040 COLOR= BG
2050 GOSUB 10200
2060 REM * IL TOPO *
2070 RC = 8
2080 R0 = PDL (0) / 6
2090 IF R0 > 39 THEN R0 = 39
2100 R1 = PDL (1) / 6
2110 IF R1 > 39 THEN R1 = 39
2120 COLOR= RC
2130 PLOT R0,R1
2140 RX = R0
2150 RY = R1
2160 REM * IL FORMAGGIO *
2170 CC = 13
2180 CX = FN R(39)
2190 CY = FN R(39)
2200 COLOR= CC
2210 PLOT CX,CY
2220 REM * IL DRAGO *
2230 DC = 1
2240 DH = 0
2250 DT = 0
2260 DX(DH) = 19
2270 DY(DH) = 19
2275 DR = 10
2280 REM * FA IL CLICK *
2290 POKE - 16336,0
2295 POKE - 16336,0
2300 REM * LOOP PRINCIPALE *
2310 REM * FORMAGGIO *
2320 IF SC < 12 THEN 2460
2330 IF SC < 20 AND TK < > 0 THEN
2460
2340 COLOR= BG
2350 PLOT CX,CY
2360 CX = CX + SGN (CX - RX)
2370 IF CX > 39 THEN CX = 39
2380 IF CX < 0 THEN CX = 0
2400 CY = CY + SGN (CY - RY)
2410 IF CY > 39 THEN CY = 39
2420 IF CY < 0 THEN CY = 0
2430 COLOR= CC
2440 PLOT CX,CY
2450 GOTO 2470
2460 FOR T = 0 TO 15: NEXT T
2470 REM * IL TOPO *
2480 R0 = INT ( PDL (0) / 6)
2490 IF R0 > 39 THEN R0 = 39
2500 R1 = INT ( PDL (1) / 6)
2510 IF R1 > 39 THEN R1 = 39
2520 IF R0 = RX AND R1 = RY THEN
2590
2530 COLOR= BG
2540 PLOT RX,RY
2550 COLOR= RC
2560 PLOT R0,R1
2570 RX = R0
2580 RY = R1
2590 IF RX < > CX OR RY < > CY
THEN 2690

```

```

2600 SC = SC + 1
2610 VTAB 22
2620 HTAB 31
2630 PRINT SC
2640 CX = FN R(39)
2650 CY = FN R(39)
2660 POKE - 16336,0
2665 POKE - 16336,0
2670 COLOR= CC
2680 PLOT CX,CY
2690 REM * IL DRAGO *
2700 IF SC < 4 THEN 2970
2710 D0 = RX
2720 D1 = RY
2730 IF SC < 8 THEN 2760
2735 IF SC = 16 THEN DR = 5
2740 IF ABS (D0 - CX) > DR THEN
D0 = CX + SGN (D0 - CX) * D
R
2750 IF ABS (D1 - CY) > DR THEN
D1 = CY + SGN (D1 - CY) * D
R
2760 DK = DH + 1
2780 IF DK > DN THEN DK = 0
2790 AF = DK = DT
2800 NP = DX(DH) < > D0 OR DY(DH
) < > D1
2810 OC = DH = DT
2820 IF NOT AF AND (NP OR OE) THEN
2890
2830 REM * CANCELLA LA CODA *
2840 COLOR= BG
2850 IF DX(DT) = CX AND DY(DT) =
CY THEN COLOR= CC
2860 PLOT DX(DT),DY(DT)
2870 DT = DT + 1
2880 IF DT > DN THEN DT = 0
2890 REM * MUOVI LA TESTA *
2900 D2 = DX(DH)
2910 D3 = DY(DH)
2920 DH = DK
2930 DX(DH) = D2 + SGN (D0 - D2)
2940 DY(DH) = D3 + SGN (D1 - D3)
2945 IF DX(DH) = RX AND DY(DH) =
RY THEN 3100
2950 COLOR= DC
2960 PLOT DX(DH),DY(DH)
2965 GOTO 2975
2970 FOR T = 0 TO 50: NEXT T
2975 REM * TIMER *
2980 TK = TK - 1
2990 IF TK > 0 THEN 2300
3000 POKE - 16336,0: POKE - 16
336,0
3010 TG = TG - 1
3020 VTAB 22: HTAB 12: PRINT TG;
" "
3030 TK = 3
3040 IF TG > 0 THEN 2300
3050 REM * FINE DEL GIOCO *
3060 REM * TEMPO SCADUTO *
3070 CALL - 936
3080 PRINT "TEMPO SCADUTO"
3090 GOTO 3140
3100 REM * IL DRAGO HA MORSO *
3110 CALL - 936

```

```

3120 PRINT "IL DRAGO TI HA PRESO
      A ";TG;" SECONDI"
3140 PRINT "PUNTEGGIO ";SC;" , PR
      ECEDENTE MAX ";HS
3150 PRINT "PREMI UN BOTTONE QUA
      LUNQUE "
3160 IF HS < SC THEN HS = SC
3170 GOTO 1840
9000 END

```

```

10100 FOR T = 0 TO 1000: NEXT T:
      RETURN
10200 REM * PULIZIA SCHERMO *
10210 FOR I = 0 TO 39 STEP 2
10220 VLIN 0,39 AT I
10230 VLIN 0,39 AT 39 - I
10240 HLIN 0,39 AT I
10250 HLIN 0,39 AT 39 - I
10260 NEXT I
10270 RETURN

```



## Space War

La prima volta che ho visto questo giochino in TI-BASIC mi sono tornate in mente alcune vecchie storie a fumetti in cui Zio Paperone cercava di nascondere tra gli asteroidi il suo denaro, ostacolato naturalmente da Rockerduck e dai Bassotti; a questo punto gli avversari si fronteggiavano nello spazio nei rispettivi razzi e ne succedevano di tutti i colori.

Bene, chi ha inventato questo giochino deve essersi probabilmente ispirato a qualcosa di si-

mile; disporrete infatti di due astronavi, una per voi e l'altra per il vostro avversario. In mezzo ad una fascia di asteroidi dovrete abbattere il nemico prima che lui faccia lo stesso con voi! Le istruzioni incluse vi diranno esplicitamente quali tasti usare per questa avvincente battaglia interplanetaria.

Un'unica raccomandazione: occhio al carburante!

```

100 REM *****
110 REM * SPACE WAR *
120 REM *****
130 CALL CLEAR
140 CALL SCREEN(2)
150 CALL CHAR(128,"0303030303030
303")
160 CALL CHAR(129,"C0C0C0C0C0C0C
0C")
170 CALL CHAR(120,"F0F0F0F0F0F0F
0F")
180 CALL CHAR(121,"FFFFFFF")
190 CALL CHAR(152,"081C2A4908080
808")
200 CALL CHAR(153,"08080808492A1
C08")
210 CALL CHAR(154,"00080402FF020
408")
220 CALL CHAR(155,"00102040FF402
01")
230 CALL CHAR(156,"1F03050911204
08")

```

```

240 CALL CHAR(157,"F8C0A09088040
201")
250 CALL CHAR(158,"8040201108050
31F")
260 CALL CHAR(159,"0102048890A0C
0F1")
270 A1$=CHR$(157)&" "&CHR$(152)
&" "&CHR$(156)
280 A2$=CHR$(159)&" "&CHR$(153)
&" "&CHR$(158)
290 CALL COLOR(13,2,2)
300 CALL COLOR(15,2,2)
310 FOR I=1 TO 8
320 CALL COLOR(I,2,2)
330 NEXT I
340 PRINT TAB(4);"PREMI UN TASTO
"::::
350 DATA 3,6,3,4,6,1,5,6,3,6,8,1
,7,6,3,3,11,1,5,11,1,3,15,2,5,15
,2,3,19,3,7,19,3,3,23,3
360 DATA 5,24,1,7,23,3,13,14,5,9
,21,2,11,21,2,9,25,3,11,26,1,13,
27,1

```



```

1080 PRINT "SPARARVI ATTRAVERSO
E LA "
1090 PRINT "COLLISIONE CON ESSI
E'FATALE"
1100 PRINT "::TAB(9);"*MISSILI*"
::
1110 PRINT "I MISSILI POSSONO ES
SERE"
1120 PRINT "SPARATI DA OGNI ANGO
LAZIONE."
1130 PRINT "EMETTONO UN GAS CHE
PARALIZ-"
1140 PRINT "ZA OGNI OGGETTO SULL
O SCHER-"
1150 PRINT "MO,FINCHE' IL MISSIL
E NON "
1160 PRINT "ESPLODE O ESCE DALLA
VISUALE"::::::
1170 PRINT "PREMI UN TASTO"
1180 FOR I=1 TO 8
1190 CALL COLOR(I,16,2)
1200 NEXT I
1210 CALL KEY(0,T,S)
1220 IF S=0 THEN 1210
1230 CALL CLEAR
1240 FOR I=1 TO 8
1250 CALL COLOR(I,16,2)
1260 NEXT I
1270 CALL COLOR(9,5,2)
1280 CALL COLOR(10,11,2)
1290 CALL COLOR(11,16,2)
1300 CALL COLOR(14,7,2)
1310 RU$="0808081C1C3E2A22"
1320 RD$="222A3E1C1C080808"
1330 RR$="0000E0387F38E"
1340 RL$="0000071CFE1C07"
1350 RUR$="010214789C28081"
1360 RUL$="8040281E39141008"
1370 RDR$="1008289C78140201"
1380 RDL$="081014391E284080"
1390 B$="0000001818"

```

```

1400 S1$="7F7F37371F0F03"
1410 S2$="FEFEFCFCF8F0C"
1420 S3$="00030F1F3F3F7F7F"
1430 S4$="00C0F0F8FCFCFEFE"
1440 D$="995A3CFFFF3C5A99"
1450 D1$="010204081020408"
1460 D2$="8040201008040201"
1470 D3$="1818181818181818"
1480 D4$="000000FFFF"
1490 CALL CHAR(96,RU$)
1500 CALL CHAR(97,RD$)
1510 CALL CHAR(98,RR$)
1520 CALL CHAR(99,RL$)
1530 CALL CHAR(100,RUR$)
1540 CALL CHAR(101,RUL$)
1550 CALL CHAR(102,RDR$)
1560 CALL CHAR(103,RDL$)
1570 CALL CHAR(104,RU$)
1580 CALL CHAR(105,RD$)
1590 CALL CHAR(106,RR$)
1600 CALL CHAR(107,RL$)
1610 CALL CHAR(108,RUR$)
1620 CALL CHAR(109,RUL$)
1630 CALL CHAR(110,RDR$)
1640 CALL CHAR(111,RDL$)
1650 CALL CHAR(112,B$)
1660 CALL CHAR(136,D$)
1670 CALL CHAR(137,D1$)
1680 CALL CHAR(138,D2$)
1690 CALL CHAR(139,D3$)
1700 CALL CHAR(140,D4$)
1710 CALL CHAR(144,S1$)
1720 CALL CHAR(145,S2$)
1730 CALL CHAR(146,S3$)
1740 CALL CHAR(147,S4$)
1750 CALL CLEAR
1760 BPR=50
1770 YPR=50
1780 A1=0
1790 B1=0
1800 A=12
1810 B=3
1820 C=12
1830 D=30
1840 CALL HCHAR(A,B,98)
1850 CALL HCHAR(C,D,107)
1860 CALL COLOR(15,16,2)
1870 CALL HCHAR(13,16,144)
1880 CALL HCHAR(13,17,145)
1890 CALL HCHAR(12,16,146)
1900 CALL HCHAR(12,17,147)
1910 FOR I=1 TO 7
1920 RANDOMIZE
1930 SH=INT(20*RND)+3
1940 SV=INT(23*RND)+5
1950 FOR I1=0 TO 1
1960 CALL GCHAR(SH,SV+I1,P)
1970 GOSUB 6160

```



```

1980 IF V=1 THEN 2060
1990 NEXT I1
2000 FOR I1=0 TO 1
2010 CALL GCHAR(SH-1,SV+I1,P)
2020 GOSUB 6160
2030 IF V=1 THEN 2060
2040 NEXT I1
2050 GOTO 2080
2060 V=0
2070 GOTO 1920
2080 CALL HCHAR(SH,SV,144)
2090 CALL HCHAR(SH,SV+1,145)
2100 CALL HCHAR(SH-1,SV,146)
2110 CALL HCHAR(SH-1,SV+1,147)
2120 NEXT I
2130 CALL KEY(2,K1,S1)
2140 G=0
2150 V=0
2160 IF S1<>0 THEN 2220
2170 CALL KEY(1,K,S)
2180 G=0
2190 V=0
2200 IF S<>0 THEN 2450
2210 GOTO 2130
2220 IF K1<>11 THEN 2290
2230 YPR=YPR-5
2240 IF YPR<0 THEN 6220
2250 E=C
2260 F=D
2270 G=2
2280 GOTO 3440
2290 YPR=YPR-1
2300 IF YPR<0 THEN 6220
2310 CALL GCHAR(C,D,P)
2320 CALL HCHAR(C,D,32)
2330 CALL SOUND(200,-6,0,440,0)

2340 G=1
2350 IF K1+1=1 THEN 4720
2360 IF K1=5 THEN 4800
2370 IF K1=3 THEN 4880
2380 IF K1=2 THEN 4960
2390 IF K1=6 THEN 5040
2400 IF K1=4 THEN 5150
2410 IF K1=14 THEN 5260
2420 IF K1=15 THEN 5370
2430 CALL HCHAR(C,D,P)
2440 GOTO 2170
2450 IF K<>18 THEN 2520
2460 BPR=BPR-5
2470 IF BPR<0 THEN 6250
2480 E=A
2490 F=B
2500 G=1
2510 GOTO 3440
2520 BPR=BPR-1
2530 IF BPR<0 THEN 6250
2540 CALL GCHAR(A,B,P)

```

```

2550 CALL HCHAR(A,B,32)
2560 CALL SOUND(200,-6,0,440,0)

2570 G=2
2580 IF K=5 THEN 2680
2590 IF K+1=1 THEN 2760
2600 IF K=3 THEN 2840
2610 IF K=2 THEN 2920
2620 IF K=6 THEN 3000
2630 IF K=4 THEN 3110
2640 IF K=14 THEN 3220
2650 IF K=15 THEN 3330
2660 CALL HCHAR(A,B,P)
2670 GOTO 2130
2680 A=A-1
2690 IF A<>0 THEN 2710
2700 A=24
2710 CALL GCHAR(A,B,P)
2720 GOSUB 6090
2730 IF V=1 THEN 5480
2740 CALL HCHAR(A,B,96)
2750 GOTO 2130
2760 A=A+1

2770 IF A<>25 THEN 2790
2780 A=1
2790 CALL GCHAR(A,B,P)
2800 GOSUB 6090
2810 IF V=1 THEN 5480
2820 CALL HCHAR(A,B,97)
2830 GOTO 2130
2840 B=B+1
2850 IF B<>33 THEN 2870
2860 B=1
2870 CALL GCHAR(A,B,P)
2880 GOSUB 6090
2890 IF V=1 THEN 5480
2900 CALL HCHAR(A,B,98)
2910 GOTO 2130
2920 B=B-1
2930 IF B<>0 THEN 2950
2940 B=32

2950 CALL GCHAR(A,B,P)
2960 GOSUB 6090
2970 IF V=1 THEN 5480
2980 CALL HCHAR(A,B,99)
2990 GOTO 2130
3000 A=A-1
3010 B=B+1
3020 IF A<>0 THEN 3040
3030 A=24
3040 IF B<>33 THEN 3060
3050 B=1
3060 CALL GCHAR(A,B,P)
3070 GOSUB 6090
3080 IF V=1 THEN 5480
3090 CALL HCHAR(A,B,100)
3100 GOTO 2130
3110 A=A-1

```

```

3120 B=B-1
3130 IF A<>0 THEN 3150
3140 A=24
3150 IF B<>0 THEN 3170
3160 B=32
3170 CALL GCHAR(A,B,P)
3180 GOSUB 6090
3190 IF V=1 THEN 5480
3200 CALL HCHAR(A,B,101)
3210 GOTO 2130
3220 A=A+1
3230 B=B+1
3240 IF A<>25 THEN 3260
3250 A=1
3260 IF B<>33 THEN 3280
3270 B=1
3280 CALL GCHAR(A,B,P)
3290 GOSUB 6090
3300 IF V=1 THEN 5480
3310 CALL HCHAR(A,B,102)
3320 GOTO 2130
3330 A=A+1
3340 B=B-1
3350 IF A<>25 THEN 3370
3360 A=1
3370 IF B<>0 THEN 3390
3380 B=32
3390 CALL GCHAR(A,B,P)
3400 GOSUB 6090
3410 IF V=1 THEN 5480
3420 CALL HCHAR(A,B,103)
3430 GOTO 2130
3440 CALL GCHAR(E,F,Z)
3450 CALL SOUND(100,440,0,880,0)


3460 IF Z=96 THEN 3480
3470 IF Z<>104 THEN 3610
3480 FOR I=E-1 TO 1 STEP -1
3490 CALL GCHAR(I,F,P)
3500 GOSUB 6000
3510 CALL HCHAR(I,F,112)
3520 CALL HCHAR(I,F,32)
3530 IF A<>I THEN 3560
3540 IF B<>F THEN 3560
3550 GOTO 5480
3560 IF C<>I THEN 3590
3570 IF D<>F THEN 3590
3580 GOTO 5480
3590 NEXT I
3600 GOTO 2170
3610 IF Z=97 THEN 3630
3620 IF Z<>105 THEN 3760
3630 FOR I=E+1 TO 24
3640 CALL GCHAR(I,F,P)
3650 GOSUB 6000
3660 CALL HCHAR(I,F,112)
3670 CALL HCHAR(I,F,32)
3680 IF A<>I THEN 3710

```

```

3690 IF B<>F THEN 3710
3700 GOTO 5480
3710 IF C<>I THEN 3740
3720 IF D<>F THEN 3740
3730 GOTO 5480
3740 NEXT I
3750 GOTO 2170
3760 IF Z=98 THEN 3780
3770 IF Z<>106 THEN 3910
3780 FOR I=F+1 TO 32
3790 CALL GCHAR(E,I,P)
3800 GOSUB 6000
3810 CALL HCHAR(E,I,112)
3820 CALL HCHAR(E,I,32)
3830 IF A<>E THEN 3860
3840 IF B<>I THEN 3860
3850 GOTO 5480
3860 IF C<>E THEN 3890
3870 IF D<>I THEN 3890
3880 GOTO 5480
3890 NEXT I
3900 GOTO 2170
3910 IF Z=99 THEN 3930
3920 IF Z<>107 THEN 4060
3930 FOR I=F-1 TO 1 STEP -1
3940 CALL GCHAR(E,I,P)
3950 GOSUB 6000
3960 CALL HCHAR(E,I,112)
3970 CALL HCHAR(E,I,32)
3980 IF A<>E THEN 4010
3990 IF B<>I THEN 4010
4000 GOTO 5480
4010 IF C<>E THEN 4040
4020 IF D<>I THEN 4040
4030 GOTO 5480
4040 NEXT I
4050 GOTO 2170
4060 IF Z=100 THEN 4080
4070 IF Z<>108 THEN 4230
4080 FOR I=E-1 TO 1 STEP -1
4090 F=F+1
4100 IF F=33 THEN 4220
4110 CALL GCHAR(I,F,P)
4120 GOSUB 6000
4130 CALL HCHAR(I,F,112)
4140 CALL HCHAR(I,F,32)
4150 IF A<>I THEN 4180
4160 IF B<>F THEN 4180
4170 GOTO 5480
4180 IF C<>I THEN 4210
4190 IF D<>F THEN 4210
4200 GOTO 5480
4210 NEXT I
4220 GOTO 2170
4230 IF Z=101 THEN 4250
4240 IF Z<>109 THEN 4400
4250 FOR I=E-1 TO 1 STEP -1
4260 F=F-1

```

**POSSIEDI  
IL TI-99/4A   
E SEI ALLA  
RICERCA DI  
SOFTWARE?**

*ma*

# Approfitta dell'offerta J. soft!

In seguito alle sempre più numerose richieste di programmi da parte dei possessori dell'home computer TI-99/4A, mettiamo a disposizione degli interessati un vasto assortimento a prezzi estremamente contenuti. L'offerta è valida fino ad esaurimento delle scorte. Prima ordini, più sei sicuro di ricevere quanto desideri.





**Programming Aids 1** - (Aiuto alla programmazione)  
Cod. DTXSX04 - L. 14.000

Mette a disposizione gli ausili per potenziare il linguaggio BASIC TI. Comprende le seguenti possibilità: "CATALOGARE DISCHI", "DISPLAY AT", "ACCEPT AT", "SCREEN PRINT (per la stampa su carta dello schermo)", "LOWERCASE (per la creazione di un set di lettere minuscole)", "2nd ASCII (per la costruzione e l'utilizzo di un secondo set di caratteri)", "CHARDEF (per rea-

lizzare e facilitare la definizione dei caratteri)".

Configurazione richiesta: A  
Configurazione raccomandata: D + E

**Blackjack e Poker**  
Cod. DTXMX04 - L. 25.000

Partite di carte simulate col computer, che vi consentono di puntare le somme dei vostri desideri. Possono partecipare fino a quattro giocatori

**Hunt the Wumpus** (Caccia la Mostro)  
Cod. DTXMA 12 - L. 25.000

Una caccia emozionante in un dedalo di caverne e gallerie. Cercate la tana del Mostro evitando i pericoli in agguato lungo il percorso. Soppesate e vagliate attentamente gli indizi per completare questa missione pericolosa.

Configurazione raccomandata: B

**Hangman** Il carnefice)  
Cod. DTXIX06 - L. 25.000

Il giocatore cerca di scoprire la parola segreta e ogni volta che sbaglia, si avvicina maggiormente alla forca. Potete usare 200 parole programmate in inglese oppure crearne 60 nuove.

Configurazione raccomandata: A

**Othello**  
Cod. DTXMX20 - L. 35.000

Questo antico gioco di strategia impegna a fondo anche i giocatori più smalizati. Un gioco da scacchiera che si impara in pochi minuti ma che richiede... una vita per dominarlo. Per strategie di tutte le età.

**Tombstone City: 21] secolo**  
Cod. DTXMX28 - L. 25.000

Vi trovate in una città fantasma del Far West nel XXI secolo, minacciata da un'orda di invasori verdi. Avanzate con la vostra pattuglia su carri coperti tipo Far West, cercando di arrestare i verdi "morgs".

Gioco velocissimo per una persona, Tombstone City mette alla prova la Vostra abilità strategica e la prontezza di riflessi.

Configurazione raccomandata: B

**Blasto**  
Cod. DTXMX05 - L. 25.000

Un carro armato (o due, a seconda del numero dei giocatori) entrano in lizza per far saltare un campo minato, evitando nel contempo i tiri dell'avversario. Una corsa velocissima contro il cronometro allo scopo di colpire il massimo numero di mine. Attenti a non colpire le mine se siete troppo vicini, altrimenti dovrete ricominciare da capo.

Configurazione raccomandata: B

**Personal Report Generator** (Generatore rapporti e archivi personali)  
Cod. DTXTX01 - L. 68.000

Permette di creare, edit e stampare lettere e rapporti di routine con i dati ricavati da un archivio creato esclusivamente con i moduli "Gestione dati personali (cod. DTXGX01)" o "Dati Statistici".

Configurazione raccomandata: A o C  
D + E

**A - MAZE - ING**  
Cod. DTXMX03 - L. 25.000

Se vi piacciono i labirinti, ecco il gioco per voi. Opzioni diverse vi offrono una grande varietà di situazioni. Da semplici dedali alla caccia al topo. Tredici opzioni con 5.200 variazioni possibili.

Configurazione raccomandata: B

**Oldies but Goodies** (Vecchi ma buoni)  
- Gioco II  
Cod. DTXMX19 - L. 14.000

Una serie di giochi, che include Hamurabi, Hidden Paris, Peg Jump, Cerchi e croci tridimensionali e Word Safari.

Configurazione richiesta: A

**Munch Man**  
Cod. DTXMX16 - L. 45.000

Manovrate il Munch Man attraverso un dedalo e cercate di raggiungere uno stimolatore prima di venire divorati dai quattro Hoonos che stanno incalzando il Munch Man. Segnate punti collegando i passaggi con una catena continua oppure catturando gli astuti Hoonos mentre il Munch Man si ricarica con 10 stimolatori.

Configurazione raccomandata: B

**Personal Record Keeping** (Gestione dati personali)  
Cod. DTXGX01 - L. 68.000

Permette di creare, mantenere ed utilizzare un sistema d'archivio computerizzato, utile e comodo per molte applicazioni, fra cui inventario domestico, scadenze manutenzione autovettura, cartelle cliniche, oltre a un mezzo di consultazione completa per compleanni, onomastici, anniversari e altre date importanti.

Configurazione raccomandata: A o C D + E

**Market Simulation** (Simulazione di mercato)  
Cod. DTXIX07 - L. 14.000

Due giocatori si trovano in concorrenza d'affari. Scegliete quanta pubblicità fare, quanti pezzi volete produrre, ecc. ed attendete i risultati. I mutamenti economici e sociali rendono estremamente realistico questo gioco istruttivo.

Configurazione richiesta: A

**The Attack** (Attacco)  
Cod. DTXMX25 - L. 25.000

Vi attende il ruolo di capitano di un'astronave in una regione spaziale infestata da "spore" ed "estraterrestri" che vanno distrutti. Manovrate l'astronave per evitare gli extraterrestri e lanciate missili per annientare il nemico.

Configurazione raccomandata: B

Configurazione:  
A - Reistratore a cassetta e cavetto di collegamento  
B - Telecomandi a filo (coppia)  
C - Memoria a dischi comprendente una scheda comando dischi e un comando dischi  
D - Scheda interfaccia parallela o seriale RS232  
E - Stampante a matrice (o altro tipo di stampante)

**Connect Four (Filetto)**  
Cod. DTXMX08 - L. 25.000

Un impegnativo gioco di strategia verticale. I giocatori devono riuscire a collocare quattro contrassegni consecutivi in senso verticale, orizzontale o diagonale.

**Zero Zap**  
Cod. DTXMX34 - L. 18.000

Flipper computerizzato dal ritmo rapido, con effetti sonori e di illuminazione. Potete creare voi stessi il campo di gioco.

**Yahtzee**  
Cod. DTXMX33 - L. 25.000

Emozionante gioco di dadi che alla strategia abbina elementi di fortuna. I giocatori accumulano punti se escono certe combinazioni.

## **e inoltre 5 libri in lingua originale, estremamente utili per il tuo TI 99/4A:**

Editore ARC soft - autore L. Turner:

36 Texas Instruments TI99/4A Programs for Home, School & Office  
Cod. BASC002 - L. 20.000

Texas Instruments Home Computer Graphics Programs  
Cod. BASC003 - L. 22.000

101 Programming Tips & Tricks for the Texas Instruments TI 99/4A Home computer.  
Cod. BASC001 - L. 20.000

Editore GRANADA - autore G. Marshall:  
Get more from the TI99/4A  
Cod. BGRC001 - L. 15.000

Texas Instruments Home Computer Games Programs  
Cod. BASC004 - L. 20.000

**Ritaglia ed invia a J. soft il tagliando sotto riportato, debitamente compilato in ogni sua parte.**

**Spett. J. soft - Via Rosellini, 12 - 20124 Milano (MI)  
Tel. 02/6888228-6880841-6880842-6880843-683797**

Ordino il seguente software/libri per il TI-99/4A:

cod. ....cod. ....cod. ....  
cod. ....cod. ....cod. ....  
cod. ....cod. ....cod. ....

per un totale di L. .... + L. 2.500 per contributo fisso di spese di spedizione.

Scelgo la seguente modalità di pagamento:

- pagherò in contrassegno al postino
- allego assegno (o contanti)
- verso l'importo sul C.C.P. n. 19445204 intestato a J. soft (allego ricevuta)

Nome .....

Cognome .....

Via ..... n. ....

CAP ..... Città ..... Prov. ....

```

4270 IF F=0 THEN 4390
4280 CALL GCHAR(I,F,P)
4290 GOSUB 6000
4300 CALL HCHAR(I,F,112)
4310 CALL HCHAR(I,F,32)
4320 IF A<>I THEN 4350
4330 IF B<>F THEN 4350
4340 GOTO 5480
4350 IF C<>I THEN 4380
4360 IF D<>F THEN 4380
4370 GOTO 5480
4380 NEXT I
4390 GOTO 2170
4400 IF Z=102 THEN 4420
4410 IF Z<>110 THEN 4570
4420 FOR I=E+1 TO 24
4430 F=F+1
4440 IF F=33 THEN 4560
4450 CALL GCHAR(I,F,P)
4460 GOSUB 6000
4470 CALL HCHAR(I,F,112)
4480 CALL HCHAR(I,F,32)
4490 IF A<>I THEN 4520
4500 IF B<>F THEN 4520
4510 GOTO 5480
4520 IF C<>I THEN 4550
4530 IF D<>F THEN 4550
4540 GOTO 5480
4550 NEXT I
4560 GOTO 2170
4570 FOR I=E+1 TO 24
4580 F=F-1
4590 IF F=0 THEN 4710
4600 CALL GCHAR(I,F,P)
4610 GOSUB 6000
4620 CALL HCHAR(I,F,112)
4630 CALL HCHAR(I,F,32)
4640 IF A<>I THEN 4670
4650 IF B<>F THEN 4670
4660 GOTO 5480
4670 IF C<>I THEN 4700
4680 IF B<>F THEN 4700
4690 GOTO 5480
4700 NEXT I
4710 GOTO 2170
4720 C=C+1
4730 IF C<>25 THEN 4750
4740 C=1
4750 CALL GCHAR(C,D,P)
4760 GOSUB 6090
4770 IF V=1 THEN 5480
4780 CALL HCHAR(C,D,105)
4790 GOTO 2170
4800 C=C-1
4810 IF C<>0 THEN 4830
4820 C=24
4830 CALL GCHAR(C,D,P)
4840 GOSUB 6090

```

```

4850 IF V=1 THEN 5480
4860 CALL HCHAR(C,D,104)
4870 GOTO 2170
4880 D=D+1
4890 IF D<>33 THEN 4910
4900 D=1
4910 CALL GCHAR(C,D,P)
4920 GOSUB 6090
4930 IF V=1 THEN 5480
4940 CALL HCHAR(C,D,106)
4950 GOTO 2170
4960 D=D-1
4970 IF D<>0 THEN 4990
4980 D=32
4990 CALL GCHAR(C,D,P)
5000 GOSUB 6090
5010 IF V=1 THEN 5480
5020 CALL HCHAR(C,D,107)
5030 GOTO 2170
5040 C=C-1
5050 D=D+1
5060 IF C<>0 THEN 5080
5070 C=24
5080 IF D<>33 THEN 5100
5090 D=1
5100 CALL GCHAR(C,D,P)
5110 GOSUB 6090
5120 IF V=1 THEN 5480
5130 CALL VCHAR(C,D,108)
5140 GOTO 2170
5150 C=C-1
5160 D=D-1
5170 IF C<>0 THEN 5190
5180 C=24
5190 IF D<>0 THEN 5210
5200 D=32
5210 CALL GCHAR(C,D,P)
5220 GOSUB 6090
5230 IF V=1 THEN 5480
5240 CALL HCHAR(C,D,109)
5250 GOTO 2170
5260 C=C+1
5270 D=D+1
5280 IF C<>25 THEN 5300
5290 C=1
5300 IF D<>33 THEN 5320
5310 D=1
5320 CALL GCHAR(C,D,P)
5330 GOSUB 6090
5340 IF V=1 THEN 5480
5350 CALL HCHAR(C,D,110)
5360 GOTO 2170
5370 C=C+1
5380 D=D-1
5390 IF C<>25 THEN 5410
5400 C=1
5410 IF D<>0 THEN 5430
5420 D=32

```

```

5430 CALL GCHAR(C,D,P)
5440 GOSUB 6090
5450 IF V=1 THEN 5480
5460 CALL HCHAR(C,D,111)
5470 GOTO 2170
5480 CALL SOUND(1000,-5,0)
5490 IF G=2 THEN 5530
5500 E=C
5510 F=D

5520 GOTO 5550
5530 E=A
5540 F=B
5550 CALL HCHAR(E,F,136)
5560 CALL COLOR(14,2,2)
5570 IF F+1<33 THEN 5590
5580 A1=1
5590 IF F-1>0 THEN 5610
5600 B1=1
5610 IF E-1<1 THEN 5670
5620 IF B1=1 THEN 5640
5630 CALL HCHAR(E-1,F-1,138)
5640 IF A1=1 THEN 5660
5650 CALL HCHAR(E-1,F+1,137)
5660 CALL HCHAR(E-1,F,139)
5670 IF E+1>24 THEN 5730
5680 IF A1=1 THEN 5700
5690 CALL HCHAR(E+1,F+1,138)
5700 IF B1=1 THEN 5720
5710 CALL HCHAR(E+1,F-1,137)
5720 CALL HCHAR(E+1,F,139)
5730 IF B1=1 THEN 5750
5740 CALL HCHAR(E-1,F,140)
5750 IF A1=1 THEN 5770
5760 CALL HCHAR(E,F+1,140)
5770 FOR I=1 TO 10
5780 CALL COLOR(14,2,2)
5790 CALL COLOR(14,7,2)
5800 NEXT I
5810 IF PT<>9 THEN 5850
5820 PRINT "PARITA'!"
5830 PT=0
5840 GOTO 5950
5850 IF PTS>0 THEN 5870
5860 PTS=5
5870 IF G=2 THEN 5920
5880 PRINT "VINCE IL BLU!"

5890 BL=BL+PTS
5900 PTS=0
5910 GOTO 5950
5920 PRINT "VINCE IL GIALLO!"
5930 YL=YL+PTS
5940 PTS=0
5950 PRINT "SCORE: BLU "&STR$(BL)
&" , GIALLO "&STR$(YL)&". "
5960 INPUT "RITENTI?":B$
5970 IF SEG$(B$,1,1)<>"N" THEN 1
750

```

```

5980 CALL CLEAR
5990 STOP
6000 FOR X=144 TO 147
6010 IF P<>X THEN 6070
6020 IF Z>103 THEN 6050
6030 BL=BL-1
6040 GOTO 2170
6050 YL=YL-1
6060 GOTO 2170

6070 NEXT X
6080 RETURN
6090 IF P=32 THEN 6150
6100 IF P<144 THEN 6110 ELSE 613
0
6110 PT=9
6120 GOTO 6140
6130 PTS=3
6140 V=1
6150 RETURN

6160 FOR X=144 TO 147
6170 IF P<>X THEN 6200
6180 V=1
6190 RETURN
6200 NEXT X
6210 RETURN
6220 PRINT "IL GIALLO FINISCE LA
BENZINA"
6230 PTS=2
6240 GOTO 5880
6250 PRINT "IL BLU HA FINITO LA
BENZINA"
6260 PTS=2
6270 GOTO 5920
6280 END

```

### Guida per l'input dei programmi sul TI 99/4A

A causa del sistema di codifica utilizzato, possono capitare in alcune linee delle improvvise "andate a capo" delle quali, naturalmente, non va tenuto conto in fase di digitazione.

Vi potrà capitare, a volte di trovare nei listati di programmi per TI 99/4a alcuni caratteri sottolineati. La sottolineatura rappresenta una particolare codifica dei tasti di controllo: dovrete quindi premere il tasto indicato dal carattere insieme al tasto CONTROL.

Ad esempio, se trovate A dovrete battere CONTROL+A; se trovate  , dovrete battere CONTROL+SPAZIO, e così via.

# Excalibur

Raccogliete le numerosissime spade sparpagliate qua e là nei sotteranei del castello di Camelot. In ogni stanza vi sono otto spade da raccogliere entro un tempo limite, mostrato sul display in basso, oltrepassato il quale perdete una delle tre vite di cui disponete (oltre a perdere 100 punti). Inoltre in tutte le stanze, via via caratterizzate da livelli di difficoltà crescenti, potrete perdere la vita se vi lancerete da una scala o da un piano da una altezza eccessiva; o se vi scontrerete con una delle porte magiche o con la gemma fatata del mago Merlino;

o, infine, se cadrete nelle botole che interrompono la continuità dei diversi piani. Una volta raccolti tutti i premi di una stanza, la porta di uscita diverrà lampeggiante, consentendovi l'uscita verso nuove stanze e avventure.

Usate per il movimento i tasti 'Q' & 'A' per salire o scendere le scale; i tasti 'O' & 'P' per spostarvi a sinistra o a destra.

Il gioco è particolarmente curato sotto il profilo della colonna sonora: a tale scopo esso è anche dotato di due brevi routine in linguaggio macchina.

```

5 CLEAR 32499
10 POKE 23658,8: POKE 23609,10
  : BORDER 6: PAPER 0: INK 7:
  CLS
15 FOR F=0 TO 47: READ A:
  POKE USR "A"+F,A: NEXT F
20 DATA 136,255,34,255,136,255
  ,34,255
22 DATA 126,129,189,165,165,18
  9,129,126
24 DATA 7,13,25,178,100,104,24
  0,168
26 DATA 60,102,195,129,129,195
  ,102,60
28 DATA 24,60,118,251,251,118,
  60,24
30 DATA 66,66,66,126,66,66,66,
  66
35 LET HI=0: LET H$="-----":
  GO SUB 1000: LET A=1:
  GO SUB 4000
40 LET LEV=0: LET MON=15:
  LET DEL=0: LET LI=3: LET S
  C=0
45 CLS : LET TIME=100: LET SW=
  0: LET X=1: LET Y=17
50 PRINT AT 0,0; INK 5;"AAAAAA
  AAAAAAAAAAAAAAAAAAAAAAAAAA"
60 PRINT AT 18,0; INVERSE 1;"A
  AAAAAAAAAAAAAAAAAAAAAAAAAA
  AAAA"
70 FOR F=0 TO 18: PRINT AT F,0
  ; INK 5;"A": NEXT F:

```

```

PRINT AT 17,0; INK 5;"
{SG8}"
80 FOR F=4 TO 17: PRINT INK 5
;AT F,4;"A";AT F,3; INK 2;"
F": NEXT F: PRINT AT 3,3;
INK 2;"F"
85 FOR G=3 TO 12: PRINT AT G,5
; INK 2;"F": NEXT G: FOR G=
10 TO 17: PRINT AT G,6;
INK 2;"F": NEXT G
90 FOR F=0 TO 15: PRINT AT F,8
; INK 5;"A": NEXT F
95 PRINT AT 11,20; INK 5;"AAAA
A";AT 9,16; INK 4;"AAAAA";
AT 7,12; INK 5;"AAAAA"
100 PRINT AT 15,9; INK 5;"AAAAA
AAAAAAAAAAAAAAAA"
105 BRIGHT 1: PRINT AT 17,3;"C"
;AT 3,4;"C";AT 10,24;"C";
AT 8,20;"C";AT 6,16;"C";
AT 3,18;"C";AT 3,26;"C";
AT 9,30;"C": BRIGHT 0
110 FOR F=10 TO 17: PRINT AT F,
22; INK 2;"F": NEXT F:
FOR F=8 TO 14: PRINT AT F,
18; INK 2;"F": NEXT F:
FOR F=6 TO 14: PRINT AT F,
14; INK 2;"F": NEXT F
115 FOR F=0 TO 17: PRINT AT F,3
1; INK 5;"A": NEXT F
120 FOR F=3 TO 15: PRINT AT F,1
0; INK 2;"F": NEXT F
130 PRINT AT 4,11; INK 5;"AAAAA
AAAAAAAAAAAA"

```

```

140 FOR F=3 TO 15: PRINT AT F,2
    8; INK 2;"F": NEXT F:
    PRINT AT 10,29;"AA"
150 IF LEV>=1 THEN PRINT AT 14
    ,12; INK 5;"A": PRINT AT 9,
    17;" "
155 IF LEV>=2 THEN PRINT AT 3,
    16; INK 5;"A": PRINT AT 11,
    21;" ";AT 5,14; INK 2;"F";
    AT 4,14; INK 2;"F";AT 3,14;
    INK 2;"F"
160 IF LEV>3 THEN PRINT AT 17,
    7; INK 5;"A": PRINT AT 14,8
    ; INK 0;" "
200 LET D=0: PRINT AT Y,X;
    INK 7; FLASH 1;"B":
    PAUSE 10: PRINT AT Y,X;
    FLASH 1;"D": PAUSE 10
205 PRINT INK 2;AT 3,3;"F";
    AT 3,5;"F";AT 10,6;"F";AT 3
    ,10;"F";AT 6,14;"F";AT 8,18
    ;"F";AT 10,22;"F";AT 3,28;"
    F"
210 PRINT AT 21,0;"PUNTI=";SC;
    AT 21,12;"VITE=";LI;AT 21,2
    2;"TEMPO=";INT TIME;" ":
    PRINT AT 20,11; INVERSE 1;
    "LIVELLO=";LEV+1
215 IF ATTR (17,3)<70 THEN
    PRINT AT 17,3; INK 2;"F"
220 IF INKEY$="P" AND ATTR (Y,X
    +1)<>5 THEN LET X=X+1:
    PRINT AT Y,X-1;" ": IF
    ATTR (Y-1,X-1)=2 THEN
    PRINT AT Y,X-1; INK 2;"F"
230 IF INKEY$="O" AND ATTR (Y,X
    -1)<>5 THEN LET X=X-1:
    PRINT AT Y,X+1;" ": IF
    ATTR (Y-1,X+1)=2 THEN
    PRINT AT Y,X+1; INK 2;"F"
240 IF ATTR (Y-1,X)=2 AND
    INKEY$="Q" THEN LET Y=Y-1:
    BEEP .007,Y: PRINT AT Y+1,
    X; INK 2;"F"
245 IF ATTR (Y+1,X)=2 AND
    INKEY$="A" THEN LET Y=Y+1:
    BEEP .007,Y: PRINT AT Y-1,
    X; INK 2;"F"
250 IF SCREEN$ (Y+1,X)=" "
    THEN LET Y=Y+1: BEEP .05,
    Y: PRINT AT Y,X; FLASH 1;"B
    ";AT Y-1,X; FLASH 0;" ":
    LET D=D+1: GO TO 410
260 IF ATTR (Y-1,X)=20 THEN
    GO SUB 6000
270 IF ATTR (Y,X+1)=71 OR
    ATTR (Y,X-1)=71 THEN
    RANDOMIZE USR 32529: LET S

```

```

C=SC+100: LET SW=SW+1:
    PRINT AT Y,X+1;" ";AT Y,X-
    1;" ": IF ATTR (Y-1,X-1)=2
    THEN PRINT AT Y,X-1;
    INK 2;"F"
280 PRINT AT 17,MON; INK 6;"E":
    PRINT AT 17,MON-1;" ":
    IF MON=25 THEN LET MON=15
    : PRINT AT 17,25;" "
285 IF X=0 AND Y=17 AND SW=8
    THEN LET LEV=LEV+1: LET S
    C=SC+1000: GO SUB 2000:
    GO TO 45
290 IF ATTR (Y,X+1)=6 OR ATTR (
    Y,X-1)=6 THEN GO SUB 6000
300 PRINT AT 19,0; INK DEL;
    INVERSE 1;" " EXCA
    LIBUR " "
310 IF TIME-.5<=0 THEN GO TO 6
    000
320 IF SW=8 THEN PRINT AT 17,0
    ; FLASH 1;" "
390 IF DEL=2 THEN PRINT AT 17,
    8;" ";AT 16,8;" ";AT 4,24;"
    ";AT 15,27;" ": BEEP .01,-
    10
395 IF DEL=4 THEN PRINT AT 17,
    8; PAPER 2; INK 4;"{G5}";
    AT 16,8;"{G5}";AT 4,24;"
    {G5}";AT 15,27;"{G5}":
    LET DEL=0: BEEP .01,0
400 LET TIME=TIME-.3: LET MON=M
    ON+1: LET DEL=DEL+1:
    GO TO 200
410 IF SCREEN$ (Y+1,X)<>" "
    THEN RANDOMIZE USR 32529
415 IF D>=6 THEN GO SUB 6000:
    GO TO 450
420 GO TO 250
1000 RESTORE 1000: FOR F=32500
    TO 32555: READ A: POKE F,A
    : NEXT F
1010 DATA 6,4,197,33,15,1,17,2,0
    ,229,205,181,3,225,17,16,0,
    167,237,90,125,254,255,32,2
    37,193,16,230,201
1020 DATA 0,6,1,197,33,0,3,17,1,
    0,229,205,181,3,225,17,16,0
    ,167,237,82,32,240,193,16,2
    33,201
1100 RETURN
2000 PRINT AT 7,0' PAPER 7," ","
    "," "," "
2020 PRINT AT 8,10; INVERSE 1;"B
    EN FATTO !";AT 10,6;"LIVELL
    O ";LEV;" COMPLETATO"
2030 FOR F=-1 TO -4 STEP -1:
    FOR E=1 TO 4: BEEP .1,F:

```

```

      BEEP .1,6: BEEP .1,13:
      BEEP .1,14
2040 IF INKEY$="" THEN NEXT E:
      NEXT F: GO TO 2030
2050 RETURN
3010 PRINT AT 9,4; FLASH 1;"LA P
      ARTITA E' TERMINATA"
3015 BEEP 1,9: BEEP 1,5: BEEP .5
      ,4: BEEP .5,7: BEEP .5,5:
      BEEP .5,-3: BEEP 2,2
3020 IF SC>HI THEN INPUT "Inser
      isci il tuo nome: "; LINE H
      $: LET HI=SC: IF LEN H$>8
      THEN GO TO 3020
3030 PAUSE 100: CLS
4000 BRIGHT 1: LET A=A+1: IF A
      >=6 THEN LET A=0
4010 PRINT AT 1,0''' "IN QUESTO G
      IOCO, TU ( B ) DEVI RACCOG
      LIERE TUTTE LE SPADE ( C )S
      PARGLIATE PER IL CASTELLO
      DI CAMELOT."
4020 PRINT INK A+3"DISPONI DI
      3 VITE. RICORDA PERO'CHE IN
      OGNI STANZA C'E' UN TEMPOL
      IMITE PER LA RACCOLTA DELL

```

```

      E SPADE, TRASCORSO IL QUA
      LE PERDI UNA VITA. INOLTRE
      DEVI EVITARE ANCHE LA GEMM
      A FATATA ( E ) E LEPORTE MA
      GICHE CHE MERLINO HA PIA
      ZZATO IN GIRO PER OSTACOLAR
      TI"
4025 PRINT INK A;"USA: Q...SU
      A...GIU' O.
      ..SINISTRA P...DESTRA "
4030 PRINT AT 21,0; INK A+3;"REC
      ORD=";HI,"DI ";H$
4100 PRINT AT 1,1; INVERSE 1;
      INK A;" EXCALIBUR
4110 RANDOMIZE USR 32500:
      RANDOMIZE USR 32529: IF
      INKEY$="" THEN GO TO 4000
4120 BRIGHT 0: CLS : GO TO 40
6000 FOR F=1 TO 4: RANDOMIZE
      USR 32500: NEXT F: LET SC=S
      C-100: LET LI=LI-1: LET SW=
      0
6010 IF LI=0 THEN GO SUB 3010
6020 CLS : GO TO 45
9999 BORDER 7: PAPER 7: INK 0:
      CLS

```



## Breakcheese

16/48K

Chi riteneva che 'breakout' fosse finito non ha certo fatto i conti con la fantasia di molti programmatori. Ecco da ogni dove apparire continuamente nuove versioni di tale gioco, migliorate, arricchite, talvolta solo lontanamente imparentate con il vetusto ma ancora autorevole capostipite. 'Topi & formaggio' è una piacevole rielaborazione dei breakout, e non solo perché sostituisce alla tradizionale pallina e al solido muro di mattoni dei voraci topolini e dei pezzettini di formaggio. In effetti il programma consente di scegliere fra due distinte modalità di gioco, caratterizzate da logiche differenti. La prima prevede che la pallina (pardon: il topo...) rimbalzi solo sui margini più esterni

dello schermo, consentendo alle bestiole di divorare grandi quantità di formaggio. La seconda modalità fa sì che il topo rimbalzi invece sugli stessi pezzettini di cibo, costringendo il giocatore a rapidi movimenti in una estenuante corsa, al fine di impedire la caduta del topolino. Il gioco termina quando vengono persi 5 topi. È importante sottolineare che, diversamente da quanto spesso succede, in questo gioco si riescono ad abbattere tutti gli strati di formaggio, risparmiando al giocatore la frustrazione che inevitabilmente si prova quando sullo schermo resta un unico, ghignante, mattone ineliminabile!

```

1 REM *****
2 REM *   TOPI & FORMAGGIO   *
3 REM *****
4 BORDER 0: PAPER 0: INK 5:
  CLS
5 REM lettura dei dati
6 FOR n=USR "a" TO USR "e"+7:
  READ k: POKE n,k: NEXT n
7 GO SUB 9000: LET hi=0:
  LET c=0: LET s=0: LET a=1:
  LET b=1
9 REM allestimento schermo
10 BORDER 0: PAPER 0: INK 5:
  CLS
15 IF s>hi THEN LET hi=s:
  CLS
16 LET s=0: LET q=5: LET c=0
20 PLOT 8,168: DRAW 239,0
30 PLOT 8,167: DRAW INK 5;0,-
  167
35 PLOT 247,167: DRAW INK 5;0
  ,-167
45 FOR n=2 TO 8
55 PRINT AT n,2; INK 6;
  BRIGHT 1;"AAAAAAAAAAAAAAAAAA
  AAAAAAAAAAAAAA": NEXT n
59 REM loop principale
60 LET n=15: LET x=20: LET y=
  INT (RND*10)+5: LET dx=x:
  LET dy=y
65 IF y>28 OR y<3 THEN
  GO SUB 150
70 IF x<2 THEN LET a=-a
75 IF SCREEN$(x,y)<>" "
  THEN GO SUB 125
80 IF x>20 AND (y=n+1 OR y=n+2
  OR y=n+3) THEN LET a=-a:
  BEEP .008,10: LET y=y+1
85 PRINT AT dx,dy;" ";AT x,y;"
  B": LET dx=x: LET dy=y
90 PRINT AT 21,n; INK 3;
  BRIGHT 1;" CDE "
95 LET n=n+(INKEY$="0" AND n
  <=26)-(INKEY$="1" AND n>0)
100 IF INKEY$="9" THEN PAUSE 0

105 LET x=x-a: LET y=y+b
110 IF x>21 THEN GO TO 165
115 PRINT AT 0,2;"PUNTI=";s;
  AT 0,13;"MAX=";hi;AT 0,24;"
  TOPI=";q
120 GO TO 65
125 BEEP .008,20
130 PRINT AT dx,dy;" ";AT x,y;
  INK 5;"B"
135 IF level=2 THEN LET a=-1

```

```

140 IF x<11 THEN LET s=s+1:
  LET c=c+1
141 IF c=loop THEN GO TO 500
145 RETURN
150 BEEP .008,30
155 LET b=1-2*(y>28 OR y<2)
160 RETURN
165 PRINT AT 0,24;"TOPI=";q-1
170 LET q=q-1
171 PAUSE 2
175 FOR u=0 TO 6: FOR d=6 TO 0
  STEP -1: BEEP .009,u:
  BEEP .009,d: NEXT u:
  NEXT d
176 IF q<>0 THEN GO TO 185
179 REM nuova partita
180 IF q<1 THEN FOR z=1 TO 6:
  PRINT AT 10,3; INK 5;
  FLASH 1;" LA PARTITA E' TE
  RMINATA ";AT 12,3;" LIVELLO
  DI GIOCO (1-2)? ": FOR d=4
  0 TO 0 STEP -5: BEEP .009,d
  : NEXT d: NEXT z: LET a=1:
  LET b=1
181 IF INKEY$="1" THEN LET loo
  p=196: LET level=1: CLS :
  GO TO 10
182 IF INKEY$="2" THEN LET loo
  p=196: LET level=2: CLS :
  GO TO 10
184 GO TO 181
185 LET m=RND
190 PRINT AT 21,0;"
  "
195 LET a=1
200 LET b=-b*(m<.5)+b*(m>=.5)
205 GO TO 60
209 REM dati e inizializzazione
210 DATA 24,60,126,247,255,239,
  126,60
212 DATA 195,209,126,90,126,102
  ,126,60
214 DATA 96,99,231,255,255,231,
  96,96
216 DATA 126,255,231,165,189,23
  1,255,36
218 DATA 6,197,231,255,255,231,
  6,6
400 REM nuovo quadro
500 LET a=1: LET b=1: LET c=0:
  FOR j=1 TO 40: PRINT AT 11
  ,10; FLASH 1; INK RND*6;"NU
  OVO QUADRO": BEEP .008,j:
  NEXT j: CLS : GO TO 20
8999 REM istruzioni
9000 CLS : PRINT INVERSE 1;"
  I TOPI & IL FORMAGGIO
  "

```



```

9015 INK 4
9020 PRINT BRIGHT 1;"Scopo del
gioco e' quello di aiuta
re i topolini a mangiare
quanti piu' pezzi di formag
gio possibili. Il problema
consiste nel fatto che il
formaggio e' molto in alt
o sullo schermo, ed i topin
i non riescono a arrivarea
prenderlo da soli. Tu puoi
aiutarli lanciandoli con
una racchetta verso l'a
lto, senza naturalmente f
arli mai ricadere a terra."
9025 PRINT BRIGHT 1;"Quando i t
opi avranno divorato tutto
un mucchio di formaggio,
chiederanno ancora il tuo a
iuto per nuove mangiate. Il
gioco termina quando la
sci cadere a terra 5 topo
lini."
9030 POKE 23692,255
9040 PRINT #0; FLASH 1; BRIGHT 1
; INK 6;" PREMI UN TASTO
PER INIZIARE "
9050 IF INKEY$="" THEN GO TO 90
50
9060 IF INKEY$>="" THEN FLASH 0
: GO TO 9100
9070 GO TO 9050
9100 CLS : PRINT "' BRIGHT 1;
INK 6;"Usa i tasti :-"'"
1.....SINISTRA"'"0.....D
ESTRA"'"9.....PAUSA"'"p
er controllare la racchetta
"
9105 PRINT "' BRIGHT 1; FLASH 1
; INK 4;" LIVELLO DI GIOCO
(1-2)? "
9110 IF INKEY$="1" THEN LET loo
p=196: LET level=1:
RETURN
9115 IF INKEY$="2" THEN LET loo
p=196: LET level=2:
RETURN
9120 GO TO 9110
9995 STOP
9996 REM per SAVE su nastro
9999 SAVE "TOPI &..." LINE 1

```



## Sintetizzatore di suono

Usando la tastiera del C64 come pannello di controllo, questo programma trasforma le due file superiori di tasti nei tasti del sintetizzatore, mentre i tasti funzione controllano l'ottava e la forma d'onda. I parametri di attack, decay, sustain, release, lunghezza, filtraggio, risonanza, etc. sono regolabili premendo i tasti

opportuni come visualizzato sullo schermo. Il programma vi informerà sui valori correnti di regolazione visualizzandoli sullo schermo. Gli effetti sonori ottenibili usando questo programma sono veramente spettacolari. Provare per credere!

```

100 GOSUB 1000
102 PRINT TAB(12);"UN MOMENTO,PR
EGO..."
110 S=13*4096+1024
120 FOR I=0 TO 28 :POKE S+I,0 :N
EXT
130 DIM F(26)
140 F1=7040 :TW=2↑(1/12)
150 FOR I=1 TO 26 :F(27-I)=F1*5.
8+30 :F1=F1/TW :NEXT
160 DIM K(255)
170 K$="Q2W3ER5T6Y7UI9O0P@-*£↑"
180 FOR I=1 TO LEN(K$) :K(ASC(MI
D$(K$,I))=I :NEXT I
200 GOSUB 1200
210 FOR I=0 TO 14 STEP 7 :POKE S
+I+5,0 :POKE S+I+6,0
220 WV=32:W=1:M=2:OC=3:HB=256:Z=
0:PY=1
225 PRINT "{SU}"; TAB(12); "
{ 19 SPAZI}"
235 REM

```

```

240 FOR I=0 TO 2
245 POKE S+2+I*7,P(8) AND 255
250 POKE S+3+I*7,P(8)/256
255 NEXT I
260 POKE S+24,P(7)*16 + 15
270 POKE S+23,P(9)*16 + 7
275 POKE S+22,P(6)/16
276 POKE S+21,P(6) AND 15
280 AV = P(1) * 16 + P(2)
285 SV = P(3) * 16 + P(4)
300 GET A$:IF A$="" THEN 300
310 FR=F(K(ASC(A$)))/M :T=V*7+S
      :IF FR=Z THEN 500
315 IF PY=1 THEN V=V+1 :IF V=3 T
      HEN V=0
320 POKE T+6,Z
325 POKE T+5,Z
330 POKE T+4,0
340 POKE T,FR-HB*INT(FR/HB)
350 POKE T+1,FR/HB
360 POKE T+6,SV
365 POKE T+5,AV
370 POKE T+4,WV+1 :FOR I=1 TO 16
      0*P(5)
375 GET A$:IF A$="" THEN NEXT I
380 POKE T+4,WV :IF A$<>"" THEN
      310
385 FOR I=1 TO 1+(P(4)/2.2)↑4
390 GET A$:IF A$<>"" THEN 310
395 NEXT I :POKE S+4,Z :POKE S+1
      1,Z :POKE S+18,Z
400 GOTO 300
500 IF A$="{F1}" THEN M=1 :OC=4
      :GOTO 300
510 IF A$="{F3}" THEN M=2 :OC=3
      :GOTO 300
520 IF A$="{F5}" THEN M=4 :OC=2
      :GOTO 300
530 IF A$="{F7}" THEN M=8 :OC=1
      :GOTO 300
540 IF A$="{F2}" THEN W=0 :WV=16
      :GOTO 300
550 IF A$="{F4}" THEN W=1 :WV=32
      :GOTO 300
560 IF A$="{F6}" THEN W=2 :WV=64
      :GOTO 300
570 IF A$="{F8}" THEN W=3 :WV=12
      8 :GOTO 300
580 IF A$<>"" THEN 600
585 PY=1-PY :IF PY<>0 THEN 300
590 POKE S+11,0 :POKE S+18,0 :V=
      0
595 GOTO 300
600 N=0
610 IF A$="A" THEN N=1 :MX=15
620 IF A$="D" THEN N=2 :MX=15
630 IF A$="S" THEN N=3 :MX=15
640 IF A$="Z" THEN N=4 :MX=15

```

```

650 IF A$="L" THEN N=5 :MX=15
660 IF A$="F" THEN N=6 :MX=4095
670 IF A$="B" THEN N=7 :MX=7
680 IF A$="K" THEN N=8 :MX=4095
690 IF A$="N" THEN N=9 :MX=15
700 IF N=0 THEN 300
750 PRINT "{SU} "; P$(N); " =";
      P(N);
755 PRINT "{ 2 SPAZI}NUOVO VAL."
      ;
760 GET A$ :I=P(N) :INPUT I
770 PRINT "{SU}{ 38 SPAZI}"
780 IF (I<0) OR (I>MX) THEN PRIN
      T "{SU} MASSIMO =" ;MX; :GOTO
      755
785 P(N) = I
790 GOTO 240
1000 REM
1002 C=29
1003 POKE 53280,0:POKE53281,0
1005 PRINT"{CLR} " : PRINT " "
1007 PRINT "{ 2 SPAZI}2 3
      { 3 SPAZI}5 6 7{ 3 SPAZI}9
      0{ 3 SPAZI}- E"; TAB(C); "
      { 4 SPAZI}F1C[<S>]"
1010 PRINT " {RVS} {DES} {DES}
      B {DES} {DES} {DES} B {DES}
      {DES} B {DES} {DES}" ; TAB
      (C); "{OFF}[<A>]CF2
      { 3 SPAZI}B"
1015 PRINT " {RVS} {DES} {DES}
      B {DES} {DES} {DES} B {DES}
      {DES} B {DES} {DES}" ; TAB
      (C); "{OFF}B{ 3 SPAZI}F3C
      [<W>]"
1020 PRINT " {RVS} {DES} {DES}
      B {DES} {DES} {DES} B {DES}
      {DES} B {DES} {DES}" ; TAB
      (C); "{OFF}[<Q>]CF4
      { 3 SPAZI}B"
1030 PRINT " {RVS} B B B B B B
      B B B B B B "; TAB(C); "
      {OFF}B{ 3 SPAZI}F5C[<W>]"
1040 PRINT " {RVS}QBWBEBRBTBYBU
      BIBOBPB@B*B↑"; TAB(C); "
      {OFF}[<Q>]CF6{ 3 SPAZI}B"
1050 PRINT TAB(C); "B{ 3 SPAZI}
      F7C[<W>]"
1060 PRINT "{ 4 SPAZI}{RVS} SOLO
      / POLIFONICO {OFF}"; TAB(C
      ); "[<Q>]CF8{ 3 SPAZI}B"
1065 PRINT TAB(C); "B{ 3 SPAZI}
      OTTAVA"
1070 PRINT "{RVS}A{OFF} ATTACK
      { 5 SPAZI}{RVS}S{OFF} SUSTA
      IN"; TAB(C-4); "FORMA ONDA"
1075 PRINT " {RVS}D{OFF} DECAY

```

```

      { 6 SPAZI}{RVS}L{OFF} LUNGH
      ."
1080 PRINT "{ 2 SPAZI}{RVS}Z
      {OFF} RELEASE{ 4 SPAZI}
      {RVS}N{OFF} RISONANZA"
1082 PRINT "{ 3 SPAZI}{RVS}F
      {OFF} FILTRO{ 5 SPAZI}{RVS}
      K{OFF} FATT.PULSE"
1084 PRINT "{ 4 SPAZI}{RVS}B
      {OFF} PASSA BANDA"
1085 PRINT "{ 3 GIU' }"
1090 RETURN
1200 REM

```

```

1210 DIM P(9) :DIM P$(9)
1212 P$(1)="ATTACK" :P(1)=2
1214 P$(2)="DECAY" :P(2)=4
1216 P$(3)="SUSTAIN" :P(3)=4
1218 P$(4)="RELEASE" :P(4)=10
1220 P$(5)="LUNGH." :P(5)=1
1222 P$(6)="FILTRO" :P(6)=500
1224 P$(7)="PASSA BANDA" :P(7)=7

1226 P$(8)="FATT.PULSE" :P(8)=40
      0
1228 P$(9)="RISONANZA" :P(9)=1
1230 RETURN

```



3 K RAM

## Il paroliere

Il paroliere è un gioco di osservazione. Facendo girare il programma, ti verrà richiesto il livello di difficoltà (da 1 a 5).

Subito dopo dovrai immettere 10 parole lunghe da 3 a 8 caratteri. Fatto ciò, il computer elaborerà il puzzle e poi lo visualizzerà. In alto a destra verrà scritta la parola da cercare.

Quando riuscirai a trovarla, dovrai digitare le coordinate di partenza della parola e la sua direzione (utilizzando la tabella posta nell'angolo in basso a sinistra dello schermo).

Se la risposta data è esatta, il computer risponderà con un 'si' stampato sullo schermo, e ti darà un punteggio che è subordinato al tempo impiegato per trovare la parola.

```

2 CLR
3 A=0
4 POKE36879,8
5 A=A+1
10 PRINT"{CLR}":PRINT"{ 7 GIU' }
   { 4 DES}{GIU'}{RVS}{ 13 Q}"
11 PRINT"{RVS}{ 4 DES}Q
   { 11 SPAZI}Q"
12 PRINT"{RVS}{ 4 DES}Q PAROLIER
   E Q"
13 PRINT"{RVS}{ 4 DES}Q
   { 11 SPAZI}Q"
14 PRINT"{RVS}{ 4 DES}{ 13 Q}
   {OFF}"
20 FORX=1TO200:NEXT:PRINT"{CLR}"

25 L$="{WHT}{YEL}{GRN}{PUR}{RED}
   {CYN}":PRINTMID$(L$,A,1):IFA<
   7THEN5
70 FORX=1TOVAL(RIGHT$(TI$,2)):R=
   RND(1):NEXT

```

```

80 S=10:W=10:DIMM(S,S),W$(W),P(S
   ,S),L(W,3),F(8)
90 POKE36879,253:PRINT"{CLR}
   {GIU'}{BLU}{ 3 SPAZI}QUALE DI
   FFICOLTA'"

95 PRINT:PRINT:PRINT
100 PRINT"{PUR}DA 1(FACILE)A 5(D
   IFF){ 3 SPAZI}3{ 3 SIN}";
110 INPUTR$:X=VAL(R$):IFX<1ANDX>
   5THEN90
120 SL=9-X

130 PRINT"{BLK}{ 2 GIU' }
   { 3 SPAZI}DIGITA"W" PAROLE"
140 PRINT:PRINT"{ 4 SPAZI}LUNGHE
   DA 3 A 8"

150 PRINT:PRINT"{ 6 SPAZI}CARATT
   ERI."
170 FORX=1TOW:L(X,1)=0:L(X,2)=0:
   L(X,3)=0
180 PRINT:PRINT"{OFF}{RED}PAROLA

```

```

#";X;TAB(8);"{ 2 DES}?
{ 3 SIN}"
190 INPUTR$:Q=LEN(R$)
200 IFQ<3THENPRINTTAB(22);"{RVS}
{BLU}{SU} *{ 3 SPAZI}TROPPO
CORTA{ 3 SPAZI}* {OFF}":GOTO
180
210 IFQ>8THENPRINTTAB(22);"{RVS}
{BLU}{SU} *{ 3 SPAZI}TROPPO
LUNGA{ 3 SPAZI}* {OFF}":GOTO
180
220 X9=0:FOR Y=1TOQ:A=ASC(MID$( "*
"+R$+"*",Y+1,1))
230 IFA<65ORA>90THENX9=1:Y=Q
240 NEXT Y:IFX9=1THENPRINTTAB(22)
"{SU} * PAROLA SBAGLIATA * "
:GOTO180
250 IFX=11THENW$(X)R$+"*":GOTO29
0
260 X9=0:FOR Y=1TOX:IFQ<=LEN(W$(Y
))-1THEN280
270 FORB=XTOY+1STEP-1:W$(B)=W$(B
-1):NEXT:W$(Y)=R$+"*":X9=1:Y
=X-1
280 NEXT
290 NEXT
295 POKE36879,194
300 PRINT"{BLU}{CLR}{ 4 GIU' }
{ 22 SPAZI}HO ABBASTANZA PAR
OLE !{ 22 SPAZI}"
310 PRINT"{PUR}{ 4 GIU' }
{ 23 SPAZI}ATTENDI UN ATTIMO
...{ 23 SPAZI}"
320 PRINT"{BLK}{ 2 GIU' }
{ 22 SPAZI}STO FACENDO IL PU
ZZLE!{ 22 SPAZI}"
340 FORX=1TOS:FOR Y=1TOS:M(Y,X)=4
2:NEXT:NEXT:Q=0
360 FORX=1TOS:FOR Y=1TOS:P(Y,X)=0
:NEXT
370 NEXT:Q=Q+1:IFQ>WTHEN760
380 G=LEN(W$(Q))-2
400 X9=0:FORX=1TOS:FOR Y=1TOS:IFP
(Y,X)=0THENX9=1:X=S:Y=S
410 NEXT:NEXT:IFX9=1THEN450
430 PRINT"{CLR}QUESTA LISTA DI P
AROLE NON E' SODDISFACENTE.
440 PRINT"DIGITA UN'ALTRA LISTA.
":GOTO130
450 A=INT(S*RND(1)+1):B=INT(S*RN
D(1)+1):IFP(B,A)<>0THEN450
460 P(B,A)=1:IFM(B,A)=42THEN490
470 IFM(B,A)<>ASC(LEFT$(W$(Q),1
))THEN400
490 FORX=1TO8:F(X)=0:NEXT
500 X9=0:FORX=1TO8:IFF(X)=0THENX
9=1:X=8
510 NEXT:IFX9=0THEN400
520 D=INT(8*RND(1)+1):IFF(D)=1TH
EN520
530 F(D)=1:ONDGOTO550,590,580,62
0,610,650,640,560
550 IF(A+G)>STHEN500
560 IF(B-G)<1THEN500
570 GOTO670
580 IF(B+G)>STHEN500
590 IF(A+G)>STHEN500
600 GOTO670
610 IF(A-G)<1THEN500
620 IF(B+G)>STHEN500
630 GOTO670
640 IF(B-G)<1THEN500
650 IF(A-G)<1THEN500
670 X=A:Y=B:X9=0:FORN=2TOG+1:GOS
UB1550:IFM(Y,X)=42THEN690
680 IFM(Y,X)<>ASC(MID$(W$(Q),N,1
))THENX9=1:N=G+1
690 NEXT:X=A:Y=B:IFX9=1THEN500
710 FORN=1TOG+1:IFM(Y,X)=42THENM
(Y,X)=ASC(MID$(W$(Q),N,1))
720 GOSUB1550:NEXT
740 L(Q,1)=A-1:L(Q,2)=B-1:L(Q,3)
=D:IFQ<WTHEN360
760 FOR Y=1TOS:FORX=1TOS:IFM(Y,X)
=42THENM(Y,X)=INT(25*RND(1)+
65)
770 NEXT:NEXT:WP=0:TS=0
775 POKE36879,15
780 PRINT"{CYN}{CLR}{ 5 GIU' }
{ 8 SPAZI}FATTO!"
790 PRINT"{GRN}{ 5 GIU' }
{ 4 SPAZI}PREMI UN TASTO"
800 R$="":GETR$:IFR$=""THEN800
810 POKE36879,25
820 PRINT"{BLU}{CLR}{GIU' }
{GIU' }{ 2 SPAZI}{RVS}COLONNA
";TAB(14);"{CYN}{RVS}{SU} PA
ROLA"
860 PRINT"{BLU}{ 4 GIU' }{RVS}R
{GIU' }{SIN}I{GIU' }{SIN}G
{GIU' }{SIN}A{ 6 SU}{ 2 SIN}
{OFF}";
861 PRINT"{BLK}{ 4 DES}";
870 FORX=0TOS-1:PRINTRIGHT$(STR$(
X),1);:NEXTX:PRINT:Y=1:GOSU
B1650
880 FOR Y=1TOS:PRINT"{DES}";RIGHT
$(STR$(Y-1),1);"-";
890 FORX=1TOS:PRINTCHR$(M(Y,X));
:NEXTX
900 PRINT"_":NEXT Y:Y=0:GOSUB1650
910 PRINT"{RED}{RVS}{GIU' }
{ 3 SPAZI}7 8 1{ 3 SPAZI}"
920 PRINT"{RVS}{ 4 SPAZI}M-N

```

```

    { 4 SPAZI}":PRINT" {RVS}
    { 3 SPAZI}6*Q*2{ 3 SPAZI}"
921 PRINT" {RVS}{ 4 SPAZI}NBM
    { 4 SPAZI}":PRINT" {RVS}
    { 3 SPAZI}5 4 3{ 3 SPAZI}"
930 G=17:GOSUB1700:PRINT:PRINT"
    {SU}"TAB(12);:PRINT" {PUR}
    {RVS}{ 2 SPAZI}PUNTI
    { 2 SPAZI}":PRINTTAB(12);"
    [<J>]{ 7 SPAZI}[<L>]"
940 PRINTTAB(12);" [<J>]
    { 2 SPAZI}0{ 4 SPAZI}[<L>]"
950 PRINTTAB(12);" [<J>]
    { 7 SPAZI}[<L>]"
951 PRINTTAB(12);" [< 9 U>]":PRIN
    T" {HOME}"
960 G=2:GOSUB1700:PRINT"
    { 9 SPAZI}"
970 WP=WP+1:IFWP>WTHEN1450
980 Q=LEN(W$(WP))-1
1000 GOSUB1700:PRINTTAB(15-(Q/2)
    );LEFT$(W$(WP),Q):TI$="0000
    00"
1020 G=4:GOSUB1700:PRINTTAB(15);
    "{BLU}POS."
1025 PRINTTAB(15)" {GRN}RIG{BLK},
    {RED}COL"
1030 FORG=6TO11:GOSUB1700:
1040 PRINT" { 5 SPAZI}":NEXTG:G=6
    :GOSUB1700
1050 B$="":GETB$:IFB$=""THEN1050
1060 IFASC(B$)=13THEN1050
1070 PRINTB$;" , " ; :IFB$="0"THENB=
    0:GOTO1090
1080 B=VAL(B$):IFB<1ORB>9THENPRI
    NT" { 2 SIN}{ 2 SPAZI}
    { 2 SIN}";:GOTO1050
1090 A$="":GETA$:IFA$=""THEN1090
1100 IFASC(A$)=13THEN1090
1110 PRINTA$:IFA$="0"THENA=0:GOT
    O1140
1120 A=VAL(A$):IFA<1ORA>9THEN103
    0
1140 G=7:GOSUB1700:PRINT"DIR:" :P
    RINT:PRINTTAB(15);" {SIN}";
1150 GETD$:IFD$=""THEN1150
1160 IFASC(D$)=13THEN1150
1170 PRINT" {SU}{DES}";D$:D=VAL(D
    $):IFD<1ORD>8THEN1140
1190 WT=TI:IFB<>L(WP,2)THEN1230
1200 IFA<>L(WP,1)THEN1230
1210 IFD=L(WP,3)THEN1360
1230 X=A+1:Y=B+1:G=LEN(W$(WP))-1
    :IFM(Y,X)<>ASC(LEFT$(W$(WP)
    ,1))THEN1300
1240 X9=0:FORN=2TOG:GOSUB1550:IF
    X<1ORX>10THEN1270
1250 IFY<1ORY>10THEN1270
1260 IFM(Y,X)=ASC(MID$(W$(WP),N,
    1))THEN1280
1270 X9=1:N=G
1280 NEXTN:IFX9=0THEN1360
1300 G=6:GOSUB1700:PRINTSPC(0);:
    B$=STR$(L(WP,2)):A$=STR$(L(
    WP,1))
1310 PRINTRIGHT$(B$,LEN(B$)-1);"
    ,";RIGHT$(A$,LEN(A$)-1)
1320 G=8:GOSUB1700:PRINTSPC(1);L
    (WP,3)
1330 G=10:GOSUB1700:PRINT"↑"
1340 G=11:GOSUB1700:PRINT"J
    {RVS} NO{OFF}"
1341 G=13:GOSUB1700:PRINT" {GIU'}
    PREMI":G=13:GOSUB1700:PRINT
    "{ 2 GIU'}SPACE"
1342 QW$="":GETQW$:IFQW$=""THEN1
    342
1343 G=10:GOSUB1700:PRINT" "
1344 G=11:GOSUB1700:PRINT"
    { 5 SPAZI}"
1345 G=13:GOSUB1700:PRINT" {GIU'}
    { 7 SPAZI}":G=13:GOSUB1700:
    PRINT" { 2 GIU'}{ 5 SPAZI}"
1350 GOTO1420
1360 IFWT<(SL*60)THENWS=100:GOTO
    1390
1370 IFWT<(SL*1200)THENWS=10:GOT
    O1390
1380 WS=5+INT((SL*1200)-WT)/60
1390 G=10:GOSUB1700:PRINT"↑"
1400 G=11:GOSUB1700:PRINT" {RVS}S
    {OFF},{SIN}"WS:TS=TS+WS:FOR
    K3=1TO2000:NEXT
1420 G=19:GOSUB1700:PRINTTS
1430 GOTO960
1450 PRINT" {HOME}{ 15 GIU'}"
1460 FORX=1TO6:PRINT"
    { 12 SPAZI}":NEXTX
1470 FORG=-2TO14:GOSUB1700
1480 PRINT" { 22 SPAZI}":NEXTG
1490 FORX=1TO1500:NEXTX:PRINT"
    {CLR}"
1491 POKE36879,76
1492 PRINT" {HOME}{ 8 GIU'}{YEL}
    GIOCHI ANCORA? (S/N)"
1500 R$="":GETR$:IFR$=""THEN1500
1505 IFR$="N"THEN1520
1510 IFR$="S"THEN90
1515 IFR$<>"N"ANDR$<>"S"THEN1500

```

```

1520 PRINT "{CLR}":POKE36879,42
1535 PRINT "{CLR}":POKE36879,110
1540 PRINT "{HOME}{ 10 GIU' }{CYN}
      { 4 SPAZI}FINE DEL GIOCO":F
      ORI=1TO1000:NEXTI
1541 PRINT "{CLR}":POKE36879,27:EN
      D
1550 ONDGOTO1560,1570,1580,1590,
      1600,1610,1620,1630
1560 Y=Y-1
1570 X=X+1:RETURN
1580 X=X+1
1590 Y=Y+1:RETURN
1600 Y=Y+1

```

```

1610 X=X-1:RETURN
1620 X=X-1
1630 Y=Y-1:RETURN
1650 PRINT "{ 2 DES}";:IFY=1THENP
      RINT "[<A>]";:GOTO1670
1660 PRINT "[<Z>]";
1670 FORX=0TOS-1:PRINT"*";:NEXTX
      :IFY=1THENPRINT "[<S>]":RETU
      RN
1680 PRINT "[<X>]":RETURN
1700 PRINT "{HOME}"TAB(14);:FORX9
      =1TOG:PRINT "{BLK}{GIU' }";:N
      EXT9:RETURN

```

Per ricevere gli arretrati di **PAPER soft** compilare il seguente tagliando:

**J.soft**      **Compilare ed inviare in busta chiusa a:**  
**via Rosellini 12 - 20142 Milano - Tel. 02/6888228-683797-6880841-2-3**

Nome \_\_\_\_\_

Cognome \_\_\_\_\_

Indirizzo \_\_\_\_\_

CAP \_\_\_\_\_ Città \_\_\_\_\_ Prov. \_\_\_\_\_

Desidero ricevere i seguenti numeri arretrati di

**PAPER soft** \_\_\_\_\_



a L. 2.000 cad. per un totale di L. .... a mezzo

assegno allegato

contanti allegati

# UN PREMIO ALLA TUA FEDELTA'

ECCO LA SORPRESA CHE TI HO  
ANNUNCIATO FIN DAL NUMERO UNO. UN  
ELEGANTE E PRATICO RACCOGLITORE PER  
ARCHIVIARE ORDINATAMENTE I PRIMI 25  
NUMERI DEL TUO SETTIMANALE. SE HAI  
SEGUITO IL MIO CONSIGLIO E HAI  
RITAGLIATO E CONSERVATO I BOLLINI  
GIALLI, PUBBLICATI SU OGNI NUMERO,  
E' GIUNTO IL MOMENTO DI RICEVERE  
GRATUITAMENTE A CASA IL COMODO  
RACCOGLITORE. COMPILA IL TAGLIANDO  
PUBBLICATO SU QUESTA PAGINA, UNISCI  
I 25 BOLLINI E SPEDISCI IL TUTTO IN  
BUSTA CHIUSA A:  
J.SOFT - VIA ROSELLINI, 12 - 20124  
MILANO



Spedire a J. *soft* - Via Rosellini, 12 - 20124 MILANO

Desidero ricevere gratuitamente il raccoglitore per Paper *Soft*. Allego i  
25 bollini gialli pubblicati su i primi 25 numeri di Paper *Soft*.

Nome ..... Cognome .....

Via ..... CAP. ....

Città ..... Prov. ....

IN MILANO - VIA MASCHERONI, 14

# IL VOSTRO "NUOVISSIMO" COMPUTER SHOP

Libri e riviste di elettronica e informatica.

# CORSI CONTINUI TUTTO L'ANNO CON I MIGLIORI SOFTERISTI



IL TELEFONO È 02-437.385